

ファミリーコンピュータ™
ファミリーベーシック™

絵でわかる

ファミコンベーシック

山下利秋

地上冒険マップつき
ゼルダの伝説
必勝攻略ガイド
ひみつをあかさすヒントがぎゅしり

任天堂™
ファミリーコンピュータ

ファミコンゲーム
必勝攻略法
スーパーマリオ/ドルアーガの塔
スターフォース/スバルタンX
逆転V/テクミガチャンピオン!


新星出版社



ファミコンのゲームソフトを
一冊でわかる



誰でもわかる



ファミコンベーシック


山下列秋



ファミコン





THE NEW YORK PUBLIC LIBRARY
ASTOR LENOX TILDEN FOUNDATION
500 5TH AVENUE NEW YORK 17, N.Y.

THE NEW YORK PUBLIC LIBRARY



THE NEW YORK PUBLIC LIBRARY

ASTOR LENOX TILDEN FOUNDATION



THE NEW YORK PUBLIC LIBRARY

は じ め に



きみ、ファミコンしちゃってる？

たくさんたくさん、ファミリーコンピュータゲームで^{たの}しんでいるんだろうね。^{なが}長い^じ時間^{あそ}遊びすぎて、おとうさんやおかあさんにしかられたりしなかったかな？

ただただ、ファミコンゲームでピコピコと^{あそ}遊ぶだけじゃなくて、ファミコンのキーボードを^て手^に入れたのだから、こんどはキーボードを^{つか}使って^{あそ}遊ぶ^{ほうほう}って方法もあるよ。でも、むずかしくて、つまらないかい？

はじめてキーボードに^{むか}向^かかって、^し初^は歩^はのベーシックからひとつずつ^{すす}進^{すす}んでいくのもいいけれど、^{たの}楽しみながら、^{あそ}遊びながらプログラミングをおぼえられたら、こんないいことはないだろう。

この本は、^{ほん}楽^{たの}しく^{あそ}遊びながら、ファミコンのベーシックをなんとなくおぼえてしまうようなねらいで書かれているんだ。いろいろなプログラムがのっていて、いろいろな^{せつめい}説明^もも書^かいてあるけれど、めんどろうだったら^{せつめい}説明^{には}目^めをつぶって、プログラムだけを^{ひょうりょう}キー^を入力してRUNさせれば、それだけで^{たの}楽しいしおもしろいし、^し知らないうちにプログラミングもおぼえてしまうはずだ。がんばって^{あそ}遊^{あそ}んでね。

も く じ

パート

0

こんにちはファミコンベーシック

1. ファミコンになにをさせる? ————— 8
 - ファミコンと画面で会話を始めてみよう(9) ● ハイカイイエと答えたら **RETURN** だ(10) ● コンピュータうらないは別の方法で(11) ● ベーシックとB Gグラフィックスタート(13)
- キーボードの使い方—12
2. プログラムのセーブとロード ————— 14
 - メモリのバックアップ(15) ● カセットレコーダにセーブ(18)
3. ファミコンを電卓にしちゃう ————— 23
4. 文字や数字を画面に表示だよ ————— 24
 - 画面をきれいにそうじしてからPRINTだ(24) ● 入力ミスはよくあること。正しく訂正すればいい(26)
5. ダイレクトモードとプログラムモード ————— 28
 - BASICと表示させる3行のプログラム(29)

パート

1

マリオ、画面に出てこい

1. はじめましてぼくマリオだよ ————— 32
 - 画面の中央にマリオの出現(33)
2. 文字に重なってこんにちはマリオ ————— 35
 - 文字の上に重なったマリオ(36)
3. 変わるよ変わるいろいろマリオ ————— 40
4. 動いて逆立ち歩くよマリオ! ————— 46
 - キャラクタをごく単純に動かす(47) ● マリオを左から右に動かす(50) ● WALK 2のマリオを右向きにする(51)

- WALK 1 のマリオに逆立ちさせる(52) ● ほんとうに歩いて
いるようなマリオ(52) ● アニメキャラちゃかちゃか表示。目が
まわる(54)

パート

2

アニメキャラといっしょ

1. マリオ歩いてMOVE、MOVE ————— 58
 - 別れたあとでまたデートするマリオとレディ(63) ● たくさん
のキャラクタがいろいろに動く(68)
2. スターシップを画面に飛ばせ! ————— 71
 - スターシップを自由に飛ばせる(72) ● キャラクタが出現。
飛ぶスターシップ(75)
3. デートしたいなマリオとレディ ————— 78
 - マリオとレディがデートした!(78) ● 背景のある楽しいマ
リオとレディのデート(83)

パート

3

キャラクタと自由に遊ぼう

1. マリオが行ったり来たり ————— 88
 - 1 から 10 までの数をタテに表示させる(88) ● 画面全体を青い
■ で塗りつぶす(89) ● 青と白のシマ模様を表示させよう(89)
 - ＊マークを左から右に動かそう(90) ● マリオを F O R ~ N E
X T で右に動かす(91) ● 2 つのキャラクタをドッキング(94)
2. 数当てゲームでファミコンに挑戦 ————— 95
 - ファミコンの考えている数字はナニ?(95) ● 数当てゲームの
プログラムを読む(98)
3. 出てくる数はまるででたらめだ ————— 100
 - でたらめな数をでたらめに表示させる(100)
4. 文字のたし算はできるのかな ————— 102
 - 文字変数を使って文字列を表示させよう(103)


5. 飛^とんで行^いったりまた戻^{もど}ったり ————— 106
● *が好きな数だけ画面をかざる(107) ● マリオがどんどん下^{した}に下^さがっていく(110)
6. 左^さ右^{ゆう}に動^{うご}くかわいいレディ ————— 111
● コントローラでレディを左^さ右^{ゆう}に動^{うご}かす(111)
7. テレビの画^が面^{めん}に？が出^でたら ————— 115
● メッセージをつけておくと便利だよ(118) ● 押^おされたキーはなにかを調^{しら}べる(119) ● プログラムの読^よみ方^{かた}(121) ● スターキラー撃^{げき}墜^{つい}作^{さく}戦^{せん}(124) ● ゲームがもっと楽^{たの}しくなるBGグラフィック(126) ● バトルアタック(127)

4


BGグラフィックで絵を描いて

1. 絵^えのないゲームなんて ————— 132
● BGグラフィックを起^き動^{どう}させよう(133) ● カーソル(135) ● 座^ざ標^{ひょう}位^い置^ち(X, Y)(136) ● 描^かくテクニクを示^{しめ}すモード(137)
● ブロック遊^{あそ}びにも似^にた図^ず形^{けい}のもと(138) ● マス目^め4個^こにひとつ^{いろ}の色
2. テクニックをマスターしよう ————— 141
● SELECT (セレクト)(142) ● COPY (コピー)(143) ● MOVE (ムーブ)(144) ● CLEAR (クリア)(145) ● CHAR (キャラクタ)(145)
3. 描^かいた絵^えはとっておこうね ————— 146
● FILE (ファイル)(146)
- § それ行^いけゲームだLet's Key in! ————— 149
ゲームプログラム1 スロットキャラクタ(150)
ゲームプログラム2 マスターマインド(156)

パート



こんにちは ファミコン ベーシック






ステップ 1

ファミコンに なにをさせる？

♥ 計算^{けい さん}する？ 作曲^{さつ きょく}する？
それとも、うらないする？

ファミコンのキーボードを手^てに入^いれたきみは、きょうから一流^{いちりゅう}のパソコンистの道^{みち}をまっしぐらにGo! だ。

ファミリーベーシックのカートリッジを本^{ほん}体^{たい}に接^{せつ}続^{ぞく}したら、ま^まず電^{でん}源^{げん} ON。テレビの画^が面^{めん}にコンピュ^こータ^ーの絵^えがで^でてくるぞ。これ^{これ}をスタート画^が面^{めん}っていうんだ。どれでもいいから、キーを押^おしてみよう。画^が面^{めん}がかわ^かわって、ほら、ファミコンがきみに話^{はな}しかけてくるよ。話^{はな}しかけられたのだから、それ^{それ}に答^{こた}えてあげよう。



●ファミコンと画面で会話を始めてみよう

「アナタハ ダレデスカ? ナマエヲ イレテクダサイ」って話しかけてきたのだから、キーボードから自分の名まえを入れてから **RETURN** キーを押す。名まえは、ローマ字でもカタカナでもいいよ。

きみの名まえは? ン? マリオだって? それじゃ、「マリオ」と入れてみよう。

名まえが
めんどうだったら
なにもしないで
RETURN
だけでもいいんだ

カタカナは
カナ キーを
押してから
その文字の
キーを
押すんだよ




*ファミリーベーシックに なにができる?

ファミリーベーシックには、ベーシックのほかに、次の5つのソフトが入っているんだよ。それぞれのソフトは、いったいなにをするんだろう?

1. GAME BASIC…ゲームのプログラム作り。
2. カリキュレタボード…いろいろな計算をする。
3. ミュージックボード…作曲をする。
4. メッセージボード…伝言板や記録板の役目をする。
5. コンピュータうらない…コンピュータうらないをする。

そう、きみはまさしくマリオなのだ。ファミコンは、ちゃんと「リョウカイ」したではないか。つづいて、ファミコンからのメッセージ。



● ハイカイエと答えたなら **RETURN** だ

GAME BASIC をしたい？ したかったら「ハイ」のキー、したくなかったら「イイエ」のキーを押すんだ。

「ハイ」のキーは
F1 キー
「イイエ」のキーは
F2 キーだよ



* 機能^{きのう}を選ぶもうひとつの方法^{ほうほう}

ファミコンベーシックでなにをするかを選択^{せんたく}するとき、画面^{がめん}の「～ニ シマスカ？」の問い^とに「ハイ」「イイエ」で答えるほかに方法^{ほうほう}がもうひとつあるよ。

問い^とを受けたとき、選択^{せんたく}したい機能^{きのう}の符号^{ごう}を、直接^{ちよくせつ}キーボードから入力^{にゅうりよく}する方法だ。

- GAME BASIC → BASIC
- カリキュレータボード → CAL.
- ミュージックボード → MUS.
- メッセージボード → MES.

たとえば、ミュージックボードにしたかったら「MUS.」と入力^{にゅうりよく}して **RETURN** だ。

「イイエ」のキーを押せば、ファミコンは次のメッセージを出して聞いてくるから、自分がしたいものが出てきたときに、「ハイ」と答えればいい。

では、カリキュレータボードを動かしてみようか。

「"GAME BASIC" ニ シマスカ」の答えは **F2** キー（イイエ）だね。そして **RETURN**。

つづいて「"カリキュレータボード" ニ シマスカ」と画面に出るから、こんどは **F1** キー（ハイ）だよ。そして **RETURN** キーだ。

これでファミコンは、どんな計算でもできる状態になった。


ためしに算数の宿題でもやらせてみよう。

● コンピュータうらないは別の方法で

ただし、「コンピュータうらない」だけは、別の方法で起動しなければならないんだよ。

「～ニ シマスカ？」と画面に出たら、ファミコンにあいさつしてみよう。あいさつのことばは、「HELLO」「オハヨウ」「コンニチハ」「コンバンハ」のどれでもいい。もちろん、キーボードから、このことばを入力して

ESC キーを
押せば
スタート画面に
戻るよ



やらなければ、ファミコンには聞こえない。そして、**RETURN** キーだ。



「コンニチワ」と
にゅうりよく 入力する代わりに
F3 キーを
お 押してもいいよ

♥ ベーシックを クイックスタートさせてみよう


ベーシックの起動は、画面が「GAME BASIC」ニ シマスカ？」とたずねてきたとき、「ハイ」 **RETURN** キーとすればいい、といったけど、もっと簡単にする方法がある。この方法をクイックスタートというんだ。

電源を入れたあと、**T** キーを押しながら、ファミコンのリセットスイッチを押せばいい。これで、GAME BASIC モードに移れる。

キーボードの使い方

キーボードから、カナや英字、数字、記号などを、思い通りに入力できなければ、キーボードも宝の持ちぐされだ。

ここでは、どうすれば、どんな文字や記号が入力できるか、そのキーの使い方をいちおう書いておこう。



● ベーシックとBGグラフィックスタート

GAME BASIC モードには、
BASIC と BG GRAPHIC の2つの
機能があるから、次にはこのどちら
を使うかを、ファミコンに教えてや
るとするのが順序だよ。

キーボードの **1** キーを押せば、
BASIC がスタートだ。

キーボードの **2** キーを押せば、
BG GRAPHIC がスタートだ。

* BASIC と BG GRAPHIC

GAME BASIC モード
には2つの機能が入っ
ている。

- BASIC…ベーシックでプログラムを入力したり、プログラムを実行したりする。
- BG GRAPHIC…画面の背景に絵を描く。




キーボードの使い方

文字キーの使い方

カタカナ、英字、数字、記号を入力するのが文字キーだ。ひとつの文字キーで2種類から4種類の文字や記号が入力できるから、まず、その使い分けをはっきりおぼえておこう。

- 英字と数字……キーをそのまま押す。ただ





プログラムのセーブとロード

♥ 電源スイッチを切ってもプログラムが消えないように

これからずっと、ファミコンのキーボードをたたきながら、プログラムを入力したり、入力したプログラムを実行させたりして、長いおつきあいをしていくわけだが、プログラムを作っているときでも、実行しているときでも、「ベーシックのプログラムは、電源を切ると消えてしまう」ということを忘れないでいよう！

電源を切ってもプログラムが消えないように記録しておく方法

キーボードの使い方

し、英字と数字以外に、¥@〔;:], /などもそのまま押すだけでいい。

キーを4等分して、左側に書いてある文字や記号は、ただキーを押すだけでいいということだ。



を、プログラムのセーブというよ。

プログラムは、カセットレコーダの中^{なか}にセーブする^{ほうほう}方法と、メモリバックアップ機能^{きのう い}を生かして、一時的^{いちじてき}にプログラムを記憶^{きおく}させておく方法^{ほうほう}がある。

●メモリのバックアップ

ファミリーベーシックのカセットは、バックアップスイッチをONにしておけば、ファミリーコンピュータの電源^{でんげん}を切^きっても、メモリにあるデータ^{データ}を記憶^{きおく}させておくことができるんだ。

でも、バックアップスイッチのON、OFFは、いつやってもいいというわけではないよ。スイッチ切り替え^{きりかえ}には、タイミングがあるんだ。

■各^{かく}ボードのデータ^{データ}を残^{のこ}す

各^{かく}ボードの使用^{しよう}が終了^{しうりゆう}したら、次^{つぎ}のよう^{よう}にして、データ^{データ}を残^{のこ}すことができるよ。

操作^{そうさく}中^{ちゆう}やたらに
ON、OFFすると
へんな動作^{どうさく}を
することがあるよ




■キーボードの使い^{つか}方^{かた}

●カタカナ……[カナ]キーを押^おしてから、使^{つか}いたいカタカナの書^かいてあるキーを押^おす。

(キーの下側^{したがわ}に書^かいてある文字^{もじ})


●カタカナ小文字^{こもじ}……[カナ]キーを押^おしてから、[SHIFT]キーを押^おしながら、使^{つか}いたい字^じのキーを押^おす。



(1) **[ESC]** キーを押して、セレクト画面に戻す。

(2) 画面で「～シマスカ」と聞いてきたら、**[オ]****[ワ]****[リ]** または **[F4]** とする。

画面に「データラ
ノコシタイトキハ カ



セットノバックアップスイッチラ ONニシテクダサイ」と表示される。

(3) バックアップスイッチをONにする。

このあとは、カセットを本体から抜いても、データは記憶されているよ。

● 抜いたカセットを再び本体に接続したら

バックアップスイッチをONにして、本体からカセットを引き抜いたあと、もう一度、カセットを差し込んで、操作をつづけるときは、次のようにしよう。

キーボードの使い方

(キーの右側に書いてある文字)


● カタカナの濁音(ガギグゲゴなど)…… **[カ]**

[ナ] キーを押してカナモードにしてから、

[GRPH] キーを押しながら、使いたい字のキーを押す。




- (1) 電源^{でんげん}をONにする。
画面^{がめん}に「カセットノ
バックアップスイッ
チ OFFニ シテク
ダサイ^{ひょうじ}」と表示される。
- (2) バックアップスイ
チをOFFにする。




BASIC モードのとき

BASIC モードのとき、
それまで使^{つか}っていたベー
シックのプログラムを^{のこ}残
すには、次の方法^{つぎ ほうほう}でバッ
クアップスイッチ ON
だ。

- (1) **SYSTEM**
とする。画面^{がめん}が GAME
BASIC モードとなる。

BGグラフィックの画面の
データはメモリバックアッ
プできないよ



キーボード^{つか かた}の使い方

特殊^{とくしゅ}キー^{きー}の使い方

STOP キー、**RETURN** キー、**DEL** キー
など、ただ文字^{もじ}を入力^{にゅうりよく}するだけではなく、特
殊^{しゅ}なはたらきをするのが特殊^{とくしゅ}キーだが、ここ
では**RETURN** キーだけを説明^{せつめい}しておく。他^{ほか}
のキーはあとでね。**RETURN** キーの役目は、



(2) ファミリーベーシックのカセットのバックアップスイッチをONにする。

(3) 「3……END」を選択する。画面はスタート画面になる。

(4) ファミリーコンピュータの電源スイッチを切る。

■カセットレコーダにセーブ

せっかく苦勞して作ったプログラムは記録して残しておきたいものだ。ここでは、カセットレコーダへのプログラムの記録（セーブ）方法を書いておこう。

■キーボードとカセットレコーダの接続

キーボード裏の端子SAVE (WRITE) とカセットレコーダ端子SAVE (MIC) を接続する。同

*メモリバックアップ 使用の注意

電池の交換は、カセットを本体に差し込んでいるときに行なう。本体から抜いたまま交換すると、データは消えてしまう。

ベーシックモードを実行すると、各ボードの画面のデータは消える。

各ボードを実行すると、ベーシックプログラムのデータは消える。

■キーボードの使い方

入力の場合だ。プログラムを入力するとき、各1行を入力し終わったらかならず押す。また、入力したプログラムを実行させるときも忘れずに押す。

カーソルキーの使い方

カーソルキーには ▲ ▼ ◀ ▶ の4

RETURN キーは省略形として

☞と書くことがあるから
おぼえておいてね



じようにLOAD(READ)
とLOAD(EAR)を接続
する。

セーブした
レコーダと
同じレコーダで
ロードするんだよ



■プログラムのセーブ

メモリに記録されているプ
ログラムを、カセットテープ
に記録して保存する方法だ。
カセットテープの用意はでき
たかな？

(1) ダイレクトモードで、

SAVE "ファイルネーム"
と入力する。

*カセットレコーダの選択


カセットレコーダなら、な
んでもいいというわけではな
いよ。ファミコンベーシック
って、けっこう気むずかし屋
なのだ。

レコーダの機種によっては、
セーブ(保存)やロード(呼び
出し)ができないものもある
から、手持ちのカセットレコ
ーダがあったら、一度ためし
てみたほうがいいよ。

それがだめだったら、専用
データレコーダか、パソコン
用データレコーダを買うしか
ないな。

キーボードの使い方

個のキーがある。画面に表示されたカーソル
(■または■)の位置は、押したカーソルキ
ーの矢じるしの方角に移動する。なお、キー
を押して入力した文字や記号は、画面のカー
ソルのある位置に表示される。



(2) カセットレコーダの録音ボタンを押す。

(3) 録音状態になったら **RETUR** **N** キーを押す。プログラムの記録中は、画面に、

WRITING “ファイルネーム”

と表示されている。


SAVE “シンセイ”

WRITING シンセイ



シンセイという
プログラムの
セーブ中だ

カセットテープは
コンピュータ用の
時間の短いものが
いいんだ



(4) セーブが終了すると OK と表示される。OK と出たら、レコーダを止める。

これでプログラムのセーブはいちおう終わった。だけど、ほん

キーボードの使い方

ファンクションキーの使い方

F1 から **F8** までの8個のキーがファンクションキーだ。これらのキーには、よく使われる命令やことばが入っている。たとえば、スタートの選択画面では、「ハイ」が **F1** キー、「イイエ」が **F2** キーだったね。また、カ




とうに正確に記録されているかな？

正確に記録されているかどうか、たしかめてみよう。

(5) カセットテープを巻き戻す。

(6) LOAD? とし、レコーダの再生ボタンを押す。



LOADING “ファイルネー

ム”

OK

と表示されていたら、正しくセーブされたしるしだ。

? TP ERROR

と表示されたら、正確にセーブされていないしるしだから、音量を変えたり、テープを別なものと交換したりして、再度セーブの操作をしてみよう。

キーボードの使い方

リキュレータボードのときは、演算記号 \times \div の代わりにするよ。

ベーシックを起動したとき、ファンクションキーに入っている命令は、次のようになっている。




● **F1** LOAD (M)

● **F2** PRINT

■ プログラムのロード


カセットテープにセーブしたプログラムを、呼び出して使用する操作方法だ。

(1) LOAD “ファイルネーム” 

(2) レコーダの再生ボタンを押す。
ロードの動作が始まる。ロード中は画面に、

LOADING “ファイルネーム”
と表示されている。


(3) ロードが終了するとOKと表示されるから、レコーダを止める。



ロードのときは
音質をLOWに
しておくといいよ




* ファイルネームの省略

LOAD  とファイルネームを省略すると、どのプログラムをLOADすればいいのかわからないので、最初に見つけたプログラムをロードするんだ。

カセットにプログラムが1個だけのときは、このほうが便利だね。

■ キーボードの使い方

- F3 GOTO
- F4 CHR\$(
- F5 SPRITE
- F6 CONT(M)
- F7 LIST(M)
- F8 RUN(M)

※上で(M)とあるのは、 (RETURN キー) を押したと同じことだ。





ファミコンを 電卓にしちゃう

♥ 計算して画面に表示するのは PRINT 命令だ

ファミコンを電卓^{でんたく}にして使^{つか}っちゃおう。カリキュレータボード
なら計算^{けいさん}専門^{せんもん}だけど、ベーシックでも計算^{けいさん}はできるんだ。

25+25は？

「わかっているよ、50じゃないか」などと暗算^{あんざん}でしてしまったら
カワユクナイぞ。とにかくファミコンにやらせてみよう。

キーボードから、

PRINT 25+25


と入力^{にゅうりよく}してみよう。

PRINT 25+25

50

OK



 は RETURN キーと同じ

* ベーシック計算^{けいさん}の注意^{ちゅうい}

1. 扱^{あつか}える数^{かず}の範^{はん}圍^いは - 32768 ~ 32767 まで。この範^{はん}圍^いを超^こえると
ヘンな数^{かず}が出^でるよ。
2. 扱^{あつか}える数^{すう}字^じは整^{せい}数^{すう}だけ。割^わり
算^{ざん}で小^{しょう}数^{すう}点^{てん}が出^でたときは、小^{しょう}数^{すう}
点^{てん}以下^い切^きり捨^すてだ。



文字や数字を 画面に表示だよ

♥ ^{プ リ ン ト}PRINTは^{プ リ ン ト}PRINTでも
" " ^{か こ}で^も囲んだ文字を表示させる^{ひ ょ う じ}

キーボードがあってテレビがある。キーボードから^{にゅうりよく}入力した文字や^{すうじ}数字を、テレビに^{ひょうじ}表示させられなければ、なんのためのキーボードか、てことになるね。

テレビの画面に^{が めん}表示させる前に、画面を^{ひょうじ}そうじして、きれいにしておこう。



● ^{が めん}画面をきれいにそうじしてから^{プ リ ン ト}PRINTだ

[SHIFT] キーと **[CLR/HOME]** キーを^{どうじ}同時に^お押してみよう。

これが画面の^{が めん}そうじだ。それまで画面になにかが^{ひょうじ}表示されていたとしても、みんな^き消えてしまって、

[CLR/HOME] キー^おだけを^お押せばカーソルを^{ひかりうえ}左上の^{いち}位置に^{い どう}移動させる



これですっかりきれいになったわけだ。
それから、^でまたも出ました PRINT 命
令だ。「BASIC」という文字^もを表示^{ひょうじ}させ
てみよう。

PRINT "BASIC" 

PRINT "BASIC"


BASIC

OK

 ←カーソル

PRINT の次^{つぎ}に入^い力^{りき}した `` (ダブルクォーテーション) で囲^{かこ}まれた文字^もが、
そっくりそのまま表示^{ひょうじ}される。
PRINT 命^{めい}令^{れい}は、文字^もや数字^{すうじ}を画面^{がめん}に
表示^{ひょうじ}させる命^{めい}令^{れい}なのだ。


前^{まえ}のステップ2でも、PRINT 命^{めい}令^{れい}
が^で出てきたね。い^{ちが}ったい^{ちが}うのだ
ろう？


PRINT 25+25  としたら、50とその答^{こた}えが^{ひょうじ}されたのだ
ったね。

PRINT "25+25"

25+25

OK

 ←カーソル


CLS  でも
画面^{がめん}はきれいに
なるよ



* PRINT の省略^{しょうりゃく} 形^{けい}

PRINT は、ただ
の？で代^{だい}行^{こう}させる
こともできる。

PRINT "BASIC"
は、？を使^{つか}って、？
"BASIC" でも同^{おな}
じなのだ。

PRINT "25+25" 

と、`` で囲^{かこ}んで実行^{じっこう}させてみ
よう。

`` で囲^{かこ}まないただの数^{すう}式^{しき}
は、そのまま計^{けい}算^{さん}して、計^{けい}算^{さん}結^{けつ}

果が表示されたし、” ”で囲んだ数式は、数式そのものが表示されるのだ。この違いは、もちろんわかったね。

● 入力ミスはよくあること。正しく訂正すればいい

キーボードから入力していて「あ、まちがえちゃった！」なんてよくあること。気にしない気にしない。正しく訂正すればどうってことないんだから。では、まちがえてみよう。直してみよう。

まちがいのある行を
始めからすっきり
打ち直してもいいよ



● 1文字打ちまちがえの訂正

もちろん、A B C D E F Gときて、Gを打つつもりで、うっかりHを入力してしまったという場合だ。

カーソルキーで、カーソルをHのところに移動しよう。

A B C D E F H ←カーソル
正しいGのキーを押そう。

A B C D E F G ←カーソル

● 1文字打ち忘れた追加訂正

A B C D F G H ←カーソル

あ、DとFのあいだにEを入れ忘れた、Eを追加しなければ、という場合だ。



カーソルキーで、カーソルをFの文字の位置（Dの後ろ）に移動しよう。

←カーソル
A B C D **F** G H

INS キーを押そう。



←カーソル
A B C D F G H

カーソルの位置にあるFの字が右にずれて、カーソルの位置に1字分空白ができる。そこで、追加するEのキーを押そう。

←カーソル
A B C D E **F** G H

- 1文字余分の字を入力してしまった削除訂正



A B C D D E F G H I

Dは2個いらないから、Dを1個削除してしまおう。カーソルキーで、カーソルをEの位置（削除したいDの後ろ）に移動させよう。

←カーソル
A B C D D **E** F G H

DEL キーを押そう。



←カーソル
A B C D **E** F G H


カーソルの左の文字(D)が消え、E以下が左に1つずつ上がって、A B C D E F G Hと、ちゃんとした並びになったはずだ。



ダイレクトモードと プログラムモード

♥ プログラムの実行は RUNそしてRETURNだ

ベーシック命令には、2通りの使い方がある。ダイレクトモードとプログラムモードだ。

ダイレクトモードは、命令文を入力して、とすると同時に実行するモード。

前で勉強したPRINT "BASIC"という命令文、おぼえているかな？ **RETURN** キーを押すと同時に実行して、画面にBASICと表示したね。これがダイレクトモードなのだ。



命令を受けたら
なにも考えずに
ただそれだけを
すぐ実行するのが
ダイレクトモードだ
スカル課って
ところだね



BASICと表示させる3行のプログラム

次のプログラムを入力してみよう。

```
10 CLS
20 LOCATE 13, 12
30 PRINT "BASIC"
```

RETURN キーを押すたびに、命令文が画面に表示されるけど、ただそれだけ。PRINT "BASIC"とあるのに、BASICと表示されないではないか。というわけで、こんどはRUNと入力してだ。

画面が1度消えてから、画面のほぼ中央にBASICと表示されたね。これがプログラムモードの実行なのだよ。

BASIC



「RUN」したら
かならずで
プログラムは
動き出す



* プログラムと行番号

プログラムを作りあげる各命令文は、1行ごとに文頭に行番号をつけなければならないんだ。

コンピュータは、行番号の小さな命令文から順に実行していくから、はじめは小さな行番号から始めて、だんだん大きな行番号にしていく。


行番号は、命令の各文ごとに、たとえば、1、2、3……でも、100、200、300……でもいいけど、とりあえず最初の行を10として、20、30、40……と10きざみに付けていったほうがいいよ。なに、理由はそのうちにすぐわかるさ。

プログラムは1行ごとに行番号


プログラムモードは、ベーシックの命令文をプログラムとしていったん記憶させておき、あとでまとめて実行する。それからプログラムは、10、20、30……というように、1行の命令文ごとにかならず行番号というものが付いていることに注意してほしい。

プログラムのRUN、LIST、それから……

RUN 命令

RUN と入力して  とすれば、行番号の小さい順に1行ずつプログラムの命令を実行していく。「プログラムよ。動け!」という命令だ。

LIST 命令

メモリに記憶してあるプログラムを画面に表示しなさいという命令。入力したプログラムにまちがいはなかったかな、などと、気になるときに、LIST  とすれば、行番号順にプログラムを画面に表示させてくれるよ。

NEW 命令

それまでメモリに記憶してあるプログラムをすっかり消してしまう命令。これから新しいプログラムを入力しようというとき、古いプログラムを消してきれいにするために使われるよ。ただし、まだ必要なプログラムをNEW命令で消してしまつて、あわてたりするなよ。





パート



マリオ、
画面に出てこい





はじめまして ぼくマリオだよ

^ス^ブ^ラ^{イト}^ま^{ほう}^{つえ}
SPRITEは魔法の杖だ
^じ^{ゆう}^{しゅつ}^{げん}
キャラを自由に出現させる

アニメキャラクタマップ
Aを見てごらん。いるぞ、
いるぞ。なんともカワユク
て愉快なアニメキャラクタ
が、ソロソロいるではない
か。マリオにレディ、ペン
ペン、ニタニタ……。

こんなに楽しいアニメキ
ャラクタたちを画面に呼び
出して、^{ある}歩かせたり、^{あそ}遊ば
せたり、ケンカさせたりし
たら、ファミコンペーシッ
クが、いっぺんで好きにな
ってしまいそう。



そうだ、やるっきゃないのだ。

もんく 文句なし。次のプログラムでレッツゴー！だ。

が めん ちゆうおう しゅつげん
●画面の中央にマリオの出現

```
10 SPRITE ON
20 DEF SPRITE 0,
   (0, 1, 0, 0, 0)=CHR$
   (0)+CHR$(1)+CHR
   $(2)+CHR$(3)
30 CLS
40 SPRITE 0, 100, 100
```

プログラムにまちがいはないかな？ ゼ
ッタイまちがいないと自信マンマンなら、
そのままRUNとしてだ。

どうなったかな？ 画面のほぼ中央に、
アニメキャラクターテーブルAのマリオが表
示されたね。



プログラムは
ぎょうにゆうりよく
1行入力したら
わす
忘れずに☑だよ

にゆうりよく
入力したリストに
じしん
自信がなかったら
リスト
LIST☑で
が めん ひようじ
画面に表示させて
たしかめよう



いまはマリオが出るだけでいい

短いけど、むずかしそうなプログラムだって？ そう、そうなんだよ。たった4行のプログラムだけれど、この中にたくさんの**問題**が詰め込まれているんだ。

でもいまは、あまりプログラムのセンサクはよそう。ここではただ、このプログラムを実行すれば、マリオが出てきたゾと、おもしろがっているだけでいいよ。

このプログラムでは、**SPRITE ON**、**DEF SPRITE**、**SPRITE**の3つの命令が中心だということはわかるね。これらの命令をどう使うかは、あとでくわしく説明する機会が、きっとあるはずだよ。

* **SPRITE** かん けい めい れい 関係命令の意味

SPRITE ON

スプライト画面にキャラクターを表示させるときには、かならず実行しなければならない決まり文句だ。

DEF SPRITE

キャラクターテーブルAにあるどのアニメキャラクターを、どんな状態で表示させるかを決める。スプライトの定義というんだよ。

SPRITE

DEF SPRITEで定義したキャラクターを表示させる位置を決めて、実際に表示させる命令だ。






ステップ 2

文字に重なって こんにちはマリオ

PRINTはバックグラウンド
SPRITEはスプライト面

前に PRINT 文で、文字を画面に表示させたね。思い出してみよう。

PRINT "BASIC" 

と、こうすれば、画面に BASIC という文字が表示されたのを、おぼえているだろう。

そしてさっきは、マリオのアニメキャラクターが画面に出てきたね。「ステップ1」のプログラムの40行に、

文字などが
表示される画面と
マリオやキャラが
表示される画面は
ちがう画面だって
ホント？



40 SPRITE 0, 100, 100

とあったはずだ。

文字もキャラクタも、同じテレビの画面に表示されていたけど、ほんとうのことをいうと、文字が表示されていた画面はバックグラウンド画面、マリオが表示されていた画面はスプライト画面といって、^{おな}区別されているんだ。

同じテレビの画面なのに、どうして2つの画面があるの？っていいいのだね。まあ、その証拠をお目にかけよう。




● 文字の上に重なったマリオ ●



```
10 SPRITE ON
20 DEF SPRITE 0,
   (0, 1, 0, 0, 0)=C
   HR$(0)+CHR$(1)
   +CHR$(2)+CHR$
   (3)
30 CLS
35 LOCATE 9, 11:P
   RINT "AAAAA"
40 SPRITE 0,100,105
```

マリオのかげにAの字がかくれた！

入力したら例によって、
RUN だよ。

AAAAAという文字の^{うえ}にマリオが^{かき}重なって、しかも、
重なった部分は消えているけど、マリオは全身すっかり表示されているね。

これは、AAAAAが表示されているバックグラウンド^{めん}の上に、マリオが表示されているスプライト^{めん}面が重なっているのだよ。

これで、文字などを表示するバックグラウンド^{めん}と、アニメキャラクタを表示するスプライト^{めん}面の2枚の画面があることが、わかったね。

バックグラウンド^{めん}面

AAAAA



スプライト^{めん}面



ぜんめん 前面にスプライト^{めん}面を出す



スプライトを消すなら

とキーだよ

同時に押すんだ



SPRITE OFF

だって消えるよ

■ SPRITE ONはマリオの舞台^{ぶたい}

スプライト画面^{がめん}とバックグラウンド画面^{がめん}は、わかったよね。バックグラウンド画面^{がめん}は文字を表示させる画面、スプライト画面^{がめん}はアニメキャラクタ^{ひょうじ}が表示される画面だ。そして、SPRITE ONは、スプライト画面^{がめん}に描かれたアニメキャラクタ^かが見えるように、スプライト面^{めん}にあかりを入れる命令^{めいれい}だ。

あかりをつけるのは
SPRITE ON。
あかりがついたから
マリオのキャラクタが
見えた！

あかりを消すのは
SPRITE OFF
だよ



(バックグラウンド^{めん}面)

AAAAAAAAA

(スプライト^{めん}面)



SPRITE ON

SPRITE OFF

AAAAA Mario AAA



AAAAAAAAAAAA

スプライトの
あかりがついて
マリオが出た！

スプライトが
消えたから
バックグラウンド^{めん}面の
AA……だけが見える






ステップ 3

変わるよ変わる いろいろマリオ


DEF SPRITE文の 数字を変えれば表示も変わる

キャラクターテーブルAには、
いろいろなマリオがいるから、
そのうちのどのマリオを呼び
出すかで、マリオのキャラは
いろいろ表示させることがで
きる。これはあたりまえだ。

でも、キャラクターテーブル
Aにあるどれか1個だけのマ
リオを使うだけでも、いろい
ろなマリオが表示できるのだ。
やってみよう。




とりあえず、前に出たプログラムをちよっとだけ変えたものを、もう一度入力して、と。




```
10 SPRITE ON
20 DEF SPRITE 0,(0, 1, 0, 0, 0)=
  CHR$(0)+CHR$(1)+CHR$(2)+CHR
  $(3)
30 CLS
35 LOCATE 9, 11:PRINT "AAAAAAA"
40 SPRITE 0, 100, 105
```

と、こうだったね。

RUN  だ。

画面のほぼ中央に、
左向きのマリオくん
(WALK1)がいて、そ
うしろのAの字が、
ちょっと見えたりして
いるね。



色が変わって4通りのマリオ

プログラムの20行を見てみよう。

20 DEF SPRITE 0,(0, 1, 0, 0, 0)……とこ
うなっているね。このうち()の中の最初の0を、いろいろ変え
てみて実験だ。0、1、2、3の数字のうち、どれでもいいから、
ここにに入れてみるとどうかな？

この数字が変わるたびに、マリオの顔や洋服の色が変わったは

ずだ。マリオには、いろいろなポーズや衣裳があるのだ。

いろいろマリオ



いちばん^{うえ}上が

はいしよくばんごう^{ごう}の配色番号0のマリオ

あとは1、2、3の順^{じゆん}だ

はいしよく^か配色がかわれば

かん^か感じも変わるものだなあ！




■^も文字^じの後ろ^{うし}ろにカクレンボのマリオ

また、20^{ぎょう}行だ。

20 DEF SPRITE 0,(0,
1, 0, 0, 0).....

こんどは()の中の3番^な目^めの0を1
に^か変えてみる。どうなったかな？ A
AA.....の文字^もが表面^{ひょうめん}に出て、マリオ
はこの文字^ものかげ^{かげ}にカクレンボ。頭^{かぶ}や
あし^{あし}足がチラホラ見えるけどね。

バックグラウンド^{めん}面^{まへ}が前^でに出て
スプライト^{めん}面^{うし}が後ろ^{うしろ}にかくれたのだ



■ くるりと「マワレーミギ!」のマリオ

20 DEF SPRITE 0, (0, 1, 0, 1, 0) = CHR\$(1) + CHR\$(0) + CHR\$(3) + CHR\$(2)

さ ゆう ぎやく
左右逆のマリオ



()の中の4番目を0から1に変えるだけじゃないよ。=の次のCHR\$(ナンタラカンタラ……)とつづく()の中の順序に注目しよう。


前は(0)、(1)、(2)、(3)の順序だったはずだけど、これは(0)と(1)の順序も入れ替えて、結局(1)、(0)、(3)、(2)とするんだ。

キャラクターテーブルAの
キャラクターの向きが
左右逆に表示されたのだ

■ 逆立ちだってできるマリオ

20 DEF SPRITE 0, (0, 1, 0, 0, 1) = CHR\$(2) + CHR\$(3) + CHR\$(0) + CHR\$(1)

()の中の5番目の数字を1にして、CHR\$()の数字を(2)、(3)、(0)、(1)に順序を変えれば、どうなるかということだ。答えはわかっているね。マリオの逆立ちだ。



どこに^{ひょうじ}表示させるかを^き決める^{スプライト}SPRITE

SPRITE 0, 100, 100

0は^{ばんごう}スプライトの番号

SPRITE 0というこの0は、
スプライト^{ばんごう}番号0のスプライト
という意味だよ。そしてこの0
は、スプライトを^{ていぎ}定義するDEF
SPRITE 0……とあった0と一
致^ちしなければならないのだ。

100, 100というのは
^{がめん}画面の^{ばしょ}場所の
ことなんだよ

DEF
SPRITE 3
なら
SPRITE 3
にする



SPRITE 0, 100, 100 の100,
100は、スプライト面のX座標
(ヨコ) 100、Y座標(タテ) 100
の位置ということ。

SPRITE 0, 100, 100 をまとめ
てみよう。

DEF SPRITE で決めたキャラクタをX座標100、Y座標100の位置に表示せよということになる。

SPRITE 0, 100, 100
ヨコ100, タテ100の位置

ぼくたちは
ここに
表示される
んだよ



*LOCATE と SPRITE

LOCATE も SPRITE も、画面の中に表示させる位置を決める命令だけど、位置を指定する数値の単位は違うから、注意してくれ。

LOCATE を使うバックグラウンド面と、SPRITE を使うスプライト面では、位置を示す単位が違うからなのだ。

バックグラウンド面は、X座標(ヨコ)が0~27、Y座標(タテ)が0~23と大まかに分けられている。英数字やカナ、記号をヨコ28文字、タテ24行表示できる。これに対してスプライト面では、X座標0~255(ヨコ256ドット)、Y座標0~239(タテ240ドット)となっている。

だから、LOCATE 10, 10と同じ位置をSPRITE で指定すると、SPRITE 0, 96, 104 ぐらいになる。



動いて逆立ち 歩くよマリオ！

♠ レディちゃんゴメンナサイ マリオをどんどん動かさせ！

ここまでは、キャラクターテーブルAのアニメキャラクタを表示させる方法を、マリオくんに代表してもらって説明してきた。

SPRITE ON して、DEF SPRITE して、SPRITE するんだったね。

でも、ただ画面の中央にアニメキャラを表示させるだけなんて、なんだかとってもツマラナイみたい。もっとなんとかならないの？
って声も聞こえてくるようだ。


まあ、待て待て、アルプスを登るのだって、下から一歩一歩登っ

マリオ、マリオって
マリオばかり
取り上げるのね
私はどうなっちゃうの？

ボクが代表として
取り上げられて
いるだけだよ
レディちゃんも
ペンペンちゃんも
マリオと同じにすれば
表示されるさ



て行かなくては、頂上^{ちやうじやう}にまでたどり着^つけないんだぜ。

よし、ここではあまりクドクドと説明^{せつめい}するのはよそう。マリオを画面^{がめん}に出^だして、いろいろなことをやらせるプログラムを、どんどん出^だしていくから、どんどん入^{にゅうりよう}力^{りき}してRUN して、ナットクしてくれ！



レッツゴー、マリオ・バラエティだ。

● キャラクタをごく単純^{たんじゆん}に動か^{うご}す

最初^{さいしよ}はアニメキャラ^{あにめキャラ}抜^ぬきの練習^{れんしゆ}プログラムだ。画面^{がめん}でAの文字^{もじ}を左^{ひだり}から右^{みぎ}に動か^{うご}かしてみる。アニメキャラは出^でてこないけれど、アニメキャラ^{うご}を動か^{うご}かすのも、このプログラムが基本^{きほん}なのだとかきらめてついてこい。

フオア ネクスト
FOR~NEXTは
くり返^{かえ}しの
ルーチンだ

```
10 CLS
20 FOR I=0 TO 27
30 LOCATE I, 10:PRINT "A"
40 PAUSE 10
50 LOCATE I, 10:PRINT " "
60 NEXT
70 END
```



♠ ^{うご}動いているように^み見せるのは ^か書いて^け消して^{つぎ}次の^い位置^ちに^か書く

20 FOR I=0 TO 27

}

60 NEXT

20行のFORと60行のNEXTにはさまれた30行~50行を、0から27すなわち28回(I=0 TO 27)くり返せということだ。

なにをくり返すの？

30 LOCATE I, 10:PRINT "A"

座標位置(0, 10)、(1, 10)、(2, 10)……(27, 10)まで順に“A”という字を表示し、また、

PRINT ``は
なにもないもの(空白)を表示せよだから
表示を消すことだ

50 LOCATE I, 10:PRINT " "

(0, 10)、(1, 10)、(2, 10)……(27, 10)の位置の表示を消していけ。ただし、

ポーズは
時間かせぎの
命令だ
数値が小さくなれば
休み時間は
短くなる

40 PAUSE 10

表示させて、表示を消すまで、ちょっと時間をとれ、というわけだ。



ここでは、表示させることと、その表示が動くように見せることだが、そのテクニックは簡単だ。ヨコ（X座標）に動く場合なら、(1)ある位置に表示させる→(2)表示を消して隣の位置に表示させる……という動作をくり返せばいいのだ。

動かしたように見えるコツ

ヨーイ！

スタート位置につく
マリオ



(0, 10) の
位置に表示


ドン！

動きだすマリオ




前の表示を消
して右隣の
(1, 10) に
表示

走りつづけるマリオ



前の表示を消
して右隣の
(2, 10) に
表示

ゴールのテープを切
るマリオ



同じようにし
て、最後に、
(27, 10) に
表示

マリオを左から右に動かす

```

10 CLS
20 SPRITE ON
30 DEF SPRITE 0, (0, 1, 0, 1, 0) = CHR
   $(1) + CHR$(0) + CHR$(3) + CHR$(2)
40 FOR I = 0 TO 240 STEP 4
50 SPRITE 0, I, 100
60 PAUSE 10
70 NEXT
80 END

```



```

30 DEF SPRITE 0,
   (0, 1, 0, 1, 0) =

```

キャラクターテーブルAのマリオと上下の
向きは同じ

キャラクターテーブルAのマリオと左右の向き
を逆にする。左右同じなら 0

スプライト画面をバック
グラウンド画面の前に出
す

ふつうのアニメキャラクタ。
0 ならレーザーのキャラクタ
になる

表示させるマリオのスプライト番
号は 0



DEF SPRITE の CHR\$ (数字) の順序は、


1、0、3、2 だから、テーブルのアニメキ
ャラクタと左右の向きが逆のマリオだ。



40 FOR I=0 TO 240 STEP 4

表示するヨコの位置 (X座標) は、0 ~ 240 までだが、ただし 4
つつジャンプした位置とする。だから、0 の次は 4、8、12、
16……240 という位置に表示されていくわけだ。

(STEP 4)




● WALK 2 のマリオを右向きにする ●

10 CLS

20 SPRITE ON

30 DEF SPRITE 0, (0, 1, 0, 1, 0) = CHR
\$(5) + CHR\$(4) + CHR\$(7) + CHR\$(6)

40 SPRITE 0, 100, 100



ウ オ ー ク ^{さか だ}
WALK 1 のマリオに逆立ちさせる

```

10 CLS
20 SPRITE ON
30 DEF SPRITE 0,
    (0, 1, 0, 0, 1) =
    CHR$(2)+CHR$(3)+
    CHR$(0)+CHR$(1)
40 SPRITE 0, 100, 100
    
```



^{ある}
ほんとうに歩いているようなマリオ

```

10 CLS
20 SPRITE ON
30 DEF SPRITE 0,
    (0, 1, 0, 1, 0) =
    CHR$(1)+CHR$(0)+
    CHR$(3)+CHR$(2)
40 DEF SPRITE 1,
    (0, 1, 0, 1, 0) =
    CHR$(5)+CHR$(4)+
    CHR$(7)+CHR$(6)
50 DEF SPRITE 2,
    (0, 1, 0, 1, 0) =
    CHR$(9)+CHR$(8)+
    CHR$(11)+CHR$(10)
    
```



```

60 FOR I=0 TO 240 STEP 6
70 FOR J=0 TO 2
80 SPRITE J, I+J, 100
90 PAUSE 5
100 SPRITE J
110 NEXT: NEXT
120 END
    
```



ほんとうに足^{あし}を動か^{うご}かして歩^{ある}いているようなマリオを表示^{ひょうじ}させるなら、キャラクターテーブルAのマリオ WALK 1～3を交互^{こうご}に表示^{ひょうじ}させながら、X座標^{ざひょう}を動か^{うご}かしていけばいいのだ。

おもしろいかおもしろくないか

あそび
遊んでプログラム

アニメキャラ16個が総登場

アニメキャラ

16個表示^{ひょうじ}のポイントは
スロケラムの40行

FOR M=0 TO 15
と50行以下の

SPRITE (M,
にあるといたら
わかるかな?



動^{うご}いたり逆^{さか}立ち^だしたり歩^{ある}いたり
のマリオを表示^{ひょうじ}させるプログラム
の応用編^{おうようへん}だ。

マリオ、マリオと、いつもマリ
オだけの表示^{ひょうじ}だけでは、なんとな
くものたりないきみに贈^{おく}るプレゼ
ント。さて、オニが^で出るかジャガ
^で出るか。次^{つぎ}にのせるプログラムを
入^い力^{りき}して、やってみるっきゃない。

画面^{がめん}に登場^{とうじょう}するのは、アニメキ

キャラクターテーブルにある16個のキャラクター^{ぜんいん}全員だ。だけど、「あ、マリオが^で出た。ペンペンが^で出た。シェルクリーパーが^で出た——」などと、ゆ^{かんしやう}っくり観賞しているヒマはないよ。^{ひやう}表示された^めり消えたり、ちゃかちゃかちゃかちゃか^め目がまわる。なにしろ、この^よ世はめまぐるしいのだ。




● アニメキャラちゃかちゃか^{ひやう}表示。^め目がまわる ●

```

10  CLS
20  CGSET 1, 0
30  SPRITE ON
40  FOR M=0 TO 15
50  DEF MOVE(0)=SPRITE(M, 1, 3, 255,
    0, 0)
60  DEF MOVE(1)=SPRITE (M, 1, 3,
    255, 0, 0)
70  DEF MOVE(2)=SPRITE (M, 1, 3,
    255, 0, 0)
80  DEF MOVE(3)=SPRITE (M, 1, 3,
    255, 0, 0)
90  DEF MOVE(4)=SPRITE (M, 1, 3,

```



```




255, 0, 0)
100 DEF MOVE(5)=SPRITE (M, 1, 3,
    255, 0, 0)
110 DEF MOVE(6)=SPRITE (M, 1, 3,
    255, 0, 0)
120 DEF MOVE(7)=SPRITE (M, 1, 3,
    255, 0, 0)
130 A=RND (250)
140 B=RND (230)
150 POSITION 0, A, B
160 C=RND (250)
170 D=RND (230)
180 POSITION 1, C, D
190 E=RND (250)
200 F=RND (230)
210 POSITION 2, E, F
220 G=RND (250)
230 H=RND (230)
240 POSITION 3, G, H
250 A1=RND (250)
260 B1=RND (230)
    
```



```

270  POSITION  4, A1, B1
280  A2=RND (250)
290  B2=RND (230)
300  POSITION  5, A2, B2
310  C2=RND (250)
320  D2=RND (230)
330  POSITION  6, C2, D2
340  E2=RND (250)
350  D2=RND (230)
360  POSITION  7, E2, D2
370  MOVE  0, 1, 2, 3, 4, 5, 6, 7
380  PAUSE  20
390  FOR I=1 TO 2000:NEXT
400  ERA  0, 1, 2, 3, 4, 5, 6, 7
410  NEXT
420  END

```





パート

2

アニメキャラと いっしょ





ステップ 1

マリオ歩いて MOVE、MOVE

◆ ^デ^フ^ム^ー^ブ DEF ^ポ^ジ^シ^ョ^ン MOVE、POSITION、 ^ム^ー^ブ MOVEは連続ワザだ

キャラクターテーブルAのアニメキャラクタは、マリオだけじゃないよ。レディもスターシップもアキレスもペンペンもいる。まだまだたくさんいるから、どのアニメキャラクタと^{あそ}遊ぼうと、きみの^{じゆう}自由だ。

SPRITE ^{めいれい}命令で、アニメキャラを^{ひょうじ}表示さ



```
10  CLS
20  SPRITE  ON
30  DEF  MOVE(0)=SPRITE (0,3,1,120,
    0, 0)
40  POSITION  0, 0, 100
50  MOVE    0
60  END
```

ほうほう
せる方法をおぼえたら、こんどは、MOVE めいれい 命令でマリオを うご動かし
てみよう。

ここに あら新しく とうじょう登場したのが、
DEF MOVE、POSITION、
MOVE という めいれい命令だ。

DEF MOVE

どんなアニメキャラを、どんな
ふう うごに動かすかを き決める。

POSITION

うご動かすアニメキャラのスタート
する い位置は どこどこかだ。

MOVE

ヨーイ、ドン、スタート、うご動け！
という めいれい命令だ。

移動方向の番号

アニメキャラの い移動 ばんごう方向は番号
で示すんだよ。上下左右と右上、
しめ じょう げ さ ゆう みぎ うえ
みぎ した ひだり うえ ひだり した ほう こう
右下、左上、左下の8方向だ。動
ほう こう ばんごう ず み
く方向の番号は図を見てね。












アニメキャラは背番号付きだ

キャラクタテーブルAのアニメキャラは16個のうち、どれでも表示させることができるよ。

ひとりひとり、一匹一匹、一個一個のアニメキャラには、それぞれ0～15までの番号が決められているから、おぼえておこう。

■ アニメキャラクタの番号

0 マリオ 	4 ペンペン 	8 スターキラー 	12 レーザー 
1 レディ 	5 ファイアーボール 	9 スターシップ 	13 シェルクリーパー 
2 ファイターフライ 	6 車 	10 爆発 	14 サイドステッパ 
3 アキレス 	7 スピナー 	11 ニタニタ 	15 ニットピッカー 

DEF MOVE、POSITION、MOVE

DEF MOVE 命令

DEF MOVE (n) =

SPRITE (A, B, C, D, E, F)



0～3までの数字。4通りの配色が選べる

0か1。0はキャラが前面に表示。1はバックグラウンドが前面に出る

全移動距離。1～255で指定。指定数の4倍のドットが移動距離

移動するスピード。1～255で指定。60ドット動くのにかかる秒数だ


移動方向。0～8の数字で指定する

0～15まで。アニメキャラの番号だ

nは
0～7までだ。
キャラの
動きに
番号を
付けるんだよ



移動距離100なら
 $100 \times 4 = 400$ で
400ドット移動だ



POSITION 命令

表示させたアニメキャラのスタートする位置を決めるのが、この POSITION だ。

POSITION n, X, Y

n は動作番号だから、DEF MOVE

座標位置は
前にやったから
もう知ってるネ



0としたあとなら、ここも0で指定する。

X, Yはスタート位置の座標だ。Xはヨコの座標、Yはタテの座標だから、Xは0~240, Yは5~220の範囲で指定する。

MOVE 命令


ホントに動けよ、オイ、という命令だ。
MOVE 0なら、0という動作を始めよということだから、DEF MOVE (0)としてあったら、これと同じ0で指定するんだ。
0~7までの数字で、8種類の動作を同時にさせることができる。

POSITION

命令を

省略したら

120, 120が
スタート位置だ



8種類の動作が同時にできる

アニメキャラクタを登場させて動かしても、1個のキャラをただ動かすだけではものたりないね。当然だよ。MOVE 命令は、同時に8種類の動作までさせることができるのだから、いろいろやってみよう。

TV見ながら

たべてのんで

字を書いて

歌って

ゲームするなんて


できるかな?



わか

別れたあとでまたデートするマリオとレディ

マリオとレディを^{おな}同じ画面^{がめん}に表示^{ひょうじ}させて、同時に^{どうじ}動か^{うご}してみよう。RUNさせると、画面^{がめん}の中央^{ちゅうおう}に立^たったマリオとレディは、なにがあったのかブーン。バイバイといったんは別れたのだが、やっぱりいっしょにいないものたりなくて、またふたりでデートというしだいだ。



```

10  CLS
20  SPRITE ON
30  DEF MOVE (0) = SPRITE (0, 7, 1,
    52, 0, 0)
40  DEF MOVE (1) = SPRITE (0, 3, 1,
    52, 0, 0)
50  DEF MOVE (2) = SPRITE (1, 3, 1,
    52, 0, 0)
60  DEF MOVE (3) = SPRITE (1, 7, 1,
    52, 0, 0)
70  POSITION 0, 104, 100
80  POSITION 2, 120, 100
90  MOVE 0, 2
    
```


```

100  IF MOVE (0) = -1 OR MOVE (2)
      = -1 THEN 100
110  POSITION 1, XPOS(0), YPOS(0)
120  POSITION 3, XPOS(2), YPOS(2)
130  ERA 0, 2
140  MOVE 1, 3
150  END

```

プログラムを簡単に説明しておこう。
 90行までは説明いらすだね。もう、知っ
 ていることだもの。ただ、DEF MOVE
 文が4つ続いて、DEF MOVE(0)~
 (3)となっているところが変わっている。
 これは、画面のアニメキャラの動作が4
 種類あるということだ。

マリオが左に行って
 右に行き
 レディが右に行って
 左に行くから
 合計4つの動作だ



◆もし~ならという条件設定の IF~THEN命令

100行のIF~THEN命令は、はじめて出
てくる命令だね。もし (IF) ナニナニな
ら、そのときは (THEN) こうしなさい、
ということだ。そんなにむずかしくはない
はずだ。


IF MOVE (0) = -1 OR M
OVE (2) = -1 THEN 100
というわけだが、これは「もし、0番の動
作または2番の動作が行なわれていたら、
(-1)、100行に行け」という命令だ。

IF A=1
THEN
PRINT 'A'
というのは
もしAが1なら
Aを表示せよ

A

ということだ

0番の動作が
されていないなら
MOVE (0) = 0だ
動くは-1
動かないは0なのだ



* ERA 命令

ERA n1, n2, ...と使う。ある動作番号 (n1, n2, ...)
の動作をするアニメキャラを画面から消す命令だ。

ERA 0, 2なら0番の動作をするキャラと、2番の動作を
するキャラを消せということだ。

110行はちょっとややこしい。

110 POSITION 1, XPOS
(0), YPOS (0)

あとでくわしく説明するけど、ここでは、
動作番号1が動きだす位置 (POSITION
1) は、動作番号0番のキャラクタが表示さ
れている位置 (XPOS (0), YPOS (0))
とする、とだけいっておこう。

XPOS ()

YPOS () は


動きの方向転換に

よく使われるよ




■ POS() は動きの座標を調べる

XPOS、YPOS は、いまアニメキャラクタが、どこに表示され
ているかを調べるものだ。



マリオがいるのは
ヨコの座標の
どこだろう？

レディがいるのは
タテの座標の
どこだろう？




■ XPOS (n)

動作番号nの動きをするア
ニメキャラが、X座標のどこ
にいるかを調べる。

■ YPOS (n)

動作番号nの動きをするア


ニメキャラが、Y座標のどこにいるかを調べる。




■ ためしにプログラム追加

前のプログラムに、

```
105 PRINT XPOS (0), YPOS (0)
```

の1行を加えて、RUN  してみよう。どうなったかな？

画面には2個の数字が表示されたはずだ。左の数字は、マリオが方向転換するときのX座標値、右の数字はY座標値だ。



ヨコの座標

は0

タテの座標

は100

の位置で

マリオは

方向転換だ



● たくさんのキャラクタがいろいろに動く

マリオとレディだけでは、あまりにも単純すぎるから、このほかにペンペンやアキレスやら、いろいろなアニメキャラを加えて、いろいろに動くプログラムを載せておこう。

DEF MOVE や POSITION 文の数値を、変えて動かしてみてもおもしろいよ、きっと。



```

10  CLS
20  CGSET 1, 0
30  SPRITE ON
40  FOR M=0 TO 15
50  DEF MOVE (0) = SPRITE (M, 1, 1,
    255, 0, 0)
60  DEF MOVE (1) = SPRITE (M, 2, 1,
    255, 0, 0)
70  DEF MOVE (2) = SPRITE (M, 3, 1,
    255, 0, 0)
80  DEF MOVE (3) = SPRITE (M, 4, 1,
    255, 0, 0)
90  DEF MOVE (4) = SPRITE (M, 5, 1,
    255, 0, 0)
100 DEF MOVE (5) = SPRITE (M, 6, 1,
    255, 0, 0)
110 DEF MOVE (6) = SPRITE (M, 7, 1,
    255, 0, 0)
120 DEF MOVE (7) = SPRITE (M, 8, 1,
    255, 0, 0)
130 MOVE 0, 1, 2, 3, 4, 5, 6, 7
    
```



```

140 PAUSE 680
150 NEXT
160 FOR M=0 TO 15
170 DEF MOVE (0) =SPRITE (M, 1, 1,
    255, 0, 0)
180 DEF MOVE (1) =SPRITE (M, 2, 1,
    255, 0, 0)
190 DEF MOVE (2) =SPRITE (M, 3, 1,
    255, 0, 0)
200 DEF MOVE (3) =SPRITE (M, 4, 1,
    255, 0, 0)
210 DEF MOVE (4) =SPRITE (M, 5, 1,
    255, 0, 0)
220 DEF MOVE (5) =SPRITE (M, 6, 1,
    255, 0, 0)
230 DEF MOVE (6) =SPRITE (M, 7, 1,
    255, 0, 0)
240 DEF MOVE (7) =SPRITE (M, 8, 1,
    255, 0, 0)
250 A=RND (7)
260 B=RND (7)
270 C=RND (7)
280 D=RND (7)
290 POSITION A, 200, 100
300 POSITION B, 100, 140
310 POSITION C, 40, 40
320 POSITION D, 120, 40
330 MOVE 0, 1, 2, 3, 4, 5, 6, 7
340 PAUSE 680
350 NEXT
360 FOR M=1 TO 10
370 A1=RND (7)
380 DEF MOVE (0) =SPRITE (A1, 1,

```



```

3, 255, 0, 0)
390 A 2=RND (7)
400 DEF MOVE (1)=SPRITE (A 2, 2,
    3, 255, 0, 0)
410 A 3=RND (7)
420 DEF MOVE (2)=SPRITE (A 3, 3,
    3, 255, 0, 0)
430 A 4=RND (7)
440 DEF MOVE (3)=SPRITE (A 4, 4,
    3, 255, 0, 0)
450 A 5=RND (7)
460 DEF MOVE (4)=SPRITE (A 5, 5,
    3, 255, 0, 0)
470 A 6=RND (7)
480 DEF MOVE (5)=SPRITE (A 6, 6,
    3, 255, 0, 0)
490 A 7=RND (7)
500 DEF MOVE (6)=SPRITE (A 7, 7,
    3, 255, 0, 0)
510 A 8=RND (7)
520 DEF MOVE (7)=SPRITE (A 8, 8,
    3, 255, 0, 0)
530 E=RND (7)
540 F=RND (7)
550 G=RND (7)
560 H=RND (7)
570 POSITION E, 20, 130
580 POSITION F, 100, 50
590 POSITION G, 140, 90
600 POSITION H, 210, 60
610 MOVE ' 0, 1, 2, 3, 4, 5, 6, 7
620 PAUSE 1080
630 NEXT
640 END

```



ステップ 2

スターシップを 画面に飛ばせ！

◆コントローラで動かせば 自由にどこへでも飛んで行くよ

キャラクターAのアニメキャラ
を表示させたら、もっと自由に動かして
みようか。どうすればいいって？

コントローラを使えば、アニメキャラ
はもっと自由自在に動かせるのだ。

いままでは、アニメキャラクターA
にあるマリオやペンペン、レディな
どのキャラを、ベーシックで画面に表示
させたり、動かしたりしていたんだね。
でも、どうもその動きが遅かったり、ぎ
こちなかったり、なんとなくものたりな
い感じがしなかったかい？

こんどは、表示させたアニメキャラの
動きを、コントローラを使って操作しよ

コントローラが
キャラをあやつって
動かすんだね



っていうわけ。動きかたがもっと自由になる
から、気分もサイコー自由になるかもね。


まず、次のプログラムを入力してからだ。




★スターシップを自由に飛ばせる

```
10 CLS
20 SPRITE ON
30 FOR I=0 TO 7
40 DEF MOVE (I)=SPRITE (9, I+1,
   1, 1, 0, 0)
50 NEXT
60 MOVE 0
70 IF STRIG (0) = 8 THEN END
80 S=STICK (0)
90 IF S=0 THEN 70
100 IF S=1 THEN A=2
110 IF S=2 THEN A=6
120 IF S=4 THEN A=4
130 IF S=5 THEN A=3
140 IF S=6 THEN A=5
150 IF S=8 THEN A=0
160 IF S=9 THEN A=1
170 IF S=10 THEN A=7
180 IF MOVE (B)=-1 THEN 180
190 IF B=A THEN 240
200 ERA B
210 POSITION A, XPOS(B), YPOS(B)
220 B=A
230 MOVE A:GOTO 70
240 MOVE B:GOTO 70
```

プログラムをRUNさせてみよう。画面のほぼ中央にスターシップが表示されたね。このスターシップを動かすのは、コントローラ I だよ。



プログラムを
終了させるなら
Aボタンを押す



どんどん動かしてみてね。なに
かがわかってくるはずだ。

せっかく苦心して入力したプログラムだから、まずテープにセーブしておこうか。セーブのしかたは前に出ていたね。

十字形ボタンの押された方向調べ

次のプログラムを入力してみてね。

10 S=STICK (0)

20 PRINT S

30 GOTO 10

入力したら
RUNだ




このプログラムはなにかというと、STICK 関数を説明しているんだ。十字形ボタンをいろいろな方向に押してみよう。



ボタンを押すたびに、画面に数字が表示されるはずだ。上なら8、下なら4だ。斜めの方向にも押してみよう。




コントローラIなら
S=STICK (0)
ただ
コントローラIIなら
S=STICK (1)
だよ

つづいて短いプログラムをもうひとつ。
入力したらいつものように
RUN だ。

```
10 T=STRIG (0)
20 PRINT T
30 GOTO 10
```


RUNさせたら、コントローラI
のトリガボタンを押してみよう。前
と同じように、どのボタンが押され
たかで、違う数字が表示されるね。
STRIG 関数は、どのボタンが押され
たかを調べるものなのだ。

 表示される数値

START→1
SELECT→2
Bボタン→4
Aボタン→8


じゅう じ けい ほうこう
* 十字形ボタンの方向

じゅう じ けい お
十字形ボタンが押さ
れた方向を数値で示す
つぎ
と次のようになる。



0 はボタンが押され
ていない状態だ。
じょうたい

じゅう じ けい お
十字形ボタンの
押された方向は
前にもちよつと
出ているよ



キャラクタが出現。飛ぶスターシップ

スターシップをコントローラで自由に動かそう。スターシップが動いていないときは、7種類のキャラクタが突然出現して、右から左へ勝手に移動したりするよ。この画面に、BGグラフィックで、夜空の背景でも描けば、楽しさも立体的になるかもね。

```

10  CLS : CGSET  1, 0
20  VIEW
30  SPRITE ON
40  FOR I=0 TO 7
50  DEF MOVE (I) = SPRITE (9, I+1,
    1, 1, 0, 0)
60  NEXT
70  MOVE 0
80  IF STRIG (0) = 8 THEN END
90  S=STICK (0)
100 IF S=0 THEN 260
110 IF S=1 THEN A=2
120 IF S=2 THEN A=6
130 IF S=4 THEN A=4
140 IF S=5 THEN A=3
150 IF S=6 THEN A=5
160 IF S=8 THEN A=0
170 IF S=9 THEN A=1
180 IF S=10 THEN A=7
190 IF MOVE (B) = -1 THEN 190
200 IF B=A THEN 250
210 ERA B
220 POSITION A, XPOS(B), YPOS(B)
230 B=A
240 MOVE A : GOTO 80
250 MOVE B : GOTO 80
260 H=H+1
270 IF H>100 THEN 290
    
```

```

280 GOTO 80
290 Z=RND (7) + 1
300 IF Z=1 THEN 370
310 IF Z=2 THEN 440
320 IF Z=3 THEN 510
330 IF Z=4 THEN 570
340 IF Z=5 THEN 630
350 IF Z=6 THEN 680
360 IF Z=7 THEN 730
370 DEF SPRITE 0, (0, 1, 0, 0, 0) =
CHR$(64) + CHR$(65) + CHR$(66) + CH
R$(67)

380 FOR N=255 TO 0 STEP -4
390 SPRITE 0, N, 50
400 FOR W1=0 TO 100: NEXT
410 NEXT
420 H=0
430 GOTO 80
440 DEF SPRITE 1, (1, 1, 0, 0, 0) =
CHR$(184) + CHR$(185) + CHR$(186) +
CHR$(187)

450 FOR N=255 TO 0 STEP -4
460 SPRITE 1, N, 100
470 FOR W2=0 TO 100: NEXT
480 NEXT
490 H=0
500 GOTO 80
510 DEF SPRITE 2, (0, 1, 0, 0, 0) =
CHR$(96) + CHR$(97) + CHR$(98) + CH
R$(99)

520 FOR N1=255 TO 0 STEP -4
530 SPRITE 2, N1, 75
540 FOR W3=0 TO 100: NEXT
550 NEXT
560 H=0: GOTO 80

```

```

570  DEF SPRITE 3,(2,1,0,0,0)=
      CHR$(28)+CHR$(29)+CHR$(30)+CH
      R$(31)
580  FOR N=255 TO 0 STEP -4
590  SPRITE 3,N,125
600  FOR W4=0 TO 100:NEXT
610  NEXT
620  H=0:GOTO 80
630  DEF SPRITE 4,(2,1,0,0,0)=
      CHR$(192)+CHR$(193)+CHR$(194)+
      CHR$(195)
640  FOR N=255 TO 0 STEP -8
650  SPRITE 4,N,150
660  FOR W5=0 TO 100:NEXT
670  NEXT:H=0:GOTO 80
680  DEF SPRITE 5,(0,1,0,0,0)=
      CHR$(200)+CHR$(201)+CHR$(202)+
      CHR$(203)
690  FOR N2=255 TO 0 STEP -4
700  SPRITE 5,N2,175
710  FOR W6=0 TO 100:NEXT
720  NEXT:H=0:GOTO 80
730  DEF SPRITE 6,(0,1,0,0,0)=
      CHR$(88)+CHR$(89)+CHR$(90)+CH
      R$(91)
740  FOR N=255 TO 0 STEP -4
750  SPRITE 6,N,200
760  FOR W7=0 TO 80:NEXT
770  NEXT
780  H=0
790  GOTO 80
800  END
    
```






デートしたいな マリオとレディ

◆BGグラフィックで 背景を描いたらもっと楽しい

アニメキャラクターAのキャラを画面に登場させて、コントローラIで動かす遊びのプログラムは、前で勉強したわけだけど、その応用で、マリオとレディをデートさせてみよう。

次のプログラムを入力したら、例によって
RUN  だよ。




●マリオとレディがデートした！

```
10 CLS:CGSET 1,0
20 SPRITE ON
30 FOR I=0 TO 3
40 DEF MOVE(I)=SPRITE(0,I*2+
  1,1,1,0,0)
```

```

50 NEXT
60 FOR I=0 TO 3
70 DEF MOVE (I+4) =SPRITE (1, I*
  2+1, 1, 1, 0, 2)
80 NEXT
90 POSITION 1, 20, 120
100 MOVE 1:B=1
110 POSITION 7, 200, 120
120 MOVE 7:D=7
130 IF STRIG(0)=8 OR STRIG(1)
  =8 THEN END
140 S=STICK(0)
150 IF S=0 THEN 270
160 IF S=1 THEN A=1
170 IF S=2 THEN A=3
180 IF S=4 THEN A=2
190 IF S=8 THEN A=0
200 IF MOVE(B)=-1 THEN 260
210 IF A=B THEN 260
220 ERA B
230 POSITION A, XPOS(B), YPOS(B)

```



```

240  B=A
250  MOVE  B:GOTO  270
260  MOVE  A
270  T=STICK (1)
280  IF  T=0  THEN  130
290  IF  T=1  THEN  C=5
300  IF  T=2  THEN  C=7
310  IF  T=4  THEN  C=6
320  IF  T=8  THEN  C=4
330  IF  MOVE (D)=-1  THEN  330
340  IF  C=D  THEN  390
350  ERA  D
360  POSITION  C, XPOS(D), YPOS(D)
370  D=C
380  MOVE  C:GOTO  130
390  MOVE  D:GOTO  130

```



ふたりがうまく会えるかな？

がめんとうじょうしたの、左

側^{がわ}にマリオ、右側^{みぎがわ}にレディ
だね。このふたりはきつと
なか仲よしだよ。けどまだ、



なんとなくふたりだけでデートしたことはないんだ。このふたりをせめて、画面の中
でだけでもデートをさせてあげよう。

このマリオとレディは、ちょっと不器
用にできていて、上下左右には自由に歩け
るけど、斜めには動けないときている。そ
こらあたりはカンベンしてもらおう。

マリオを動かすのはコントローラⅠ、レ
ディを歩かせるのはコントローラⅡだよ。

あ、マリオ、横に行ったりして
斜めに行けば

レディのところまで近いのに



コントローラⅠは
マリオ

Ⅱはレディだよ

プログラムを
終わらせたいなら
どちらの
コントローラ
でもいいから
Aボタンを押す



マリオもレディも
上下左右にしか
動けないんだから
不器用なものだ



ちょっとものたりないけど、コントローラを使っていろいろな
キャラを動かすプログラムとしては、こんなところが基本形だ。
簡単にプログラムの説明をしよう。

■ “マリオとレディがデートした！” プログラムの説明

30～50行 どんなマリオをどう動かすかを
決めている。マリオの動作番号は0～3
だ。

30 FOR I=0 TO 3……Iは
0から3だよ

40 DEF MOVE (I)=……I
は動作番号なのだ

60～80行 どんなレディをどう動かすかを決めて
いる。

60 FOR I=0 TO 3……Iは0から3だけど……

70 DEF MOVE (I+4)……動作番号はIに4を加
えるから4～7だ。


90～120行 マリオとレ
ディを最初に表示させる位
置を決めて。

130行 どちらかのコント
ローラのAボタンが押さ
れたら、プログラムは終
了する。

140～260行 マリオを移動
させるきまりを作る。

270～390行 レディを移動
させるきまりを作る。

BGクラフィックで
背景を描いて
それから
マリオとレディを
デートさせれば
もっと楽しいよ



おもしろいかおもしろくないか

 あそ
遊んでプログラム

● 背景のある楽しいマリオとレディのデート ●

デートするときには、まわりに
 こうえん ^き 公園の木があったり、と ^{やま} 遠くの山が
 かすんでいたり、なにかいろい
 ろな背景 ^{はいけい} ^{かこ} 囲まれていたほうが、
 画面 ^{がめん} ^{たの} も楽しくなるね。ファミコン
 のBGグラフィックを使えば、背
 景 ^{けい} ^か が描けるから、その背景 ^{はいけい} ^{なか} の中で
 ふたりをデートさせてみよう。



```

10  CLS:CGSET 1,0
20  SPRITE ON
25  VIEW
30  FOR I=0 TO 3
40  DEF MOVE(I)=SPRITE(0,I*2+
    1,1,1,0,0)
50  NEXT
60  FOR I=0 TO 3
70  DEF MOVE(I+4)=SPRITE(1,I*
    
```

プログラムの最初は

前と同じだから

あら ^{にやうりよく} 新たに入力しなくても

ま ^り ^{よう} 前のを利用すればいい




```

      2+1, 1, 1, 0, 2)
80  NEXT
90  POSITION 1, 20, 120
100  MOVE 1:B=1
110  POSITION 7, 200, 120
120  MOVE 7:D=7
130  IF STRIG(0)=8 OR STRIG(1)
     =8 THEN END
140  S=STICK(0)
150  IF S=0 THEN 270
160  IF S=1 THEN A=1
170  IF S=2 THEN A=3
180  IF S=4 THEN A=2
190  IF S=8 THEN A=0
200  IF MOVE(B)=-1 THEN 260
210  IF A=B THEN 260
220  ERA B
230  POSITION A,XPOS(B),YPOS(B)
240  B=A
250  MOVE B:GOTO 270
260  MOVE A

```



```


270  T=STICK (1)
280  IF T=0 THEN 400
290  IF T=1 THEN C=5
300  IF T=2 THEN C=7
310  IF T=4 THEN C=6
320  IF T=8 THEN C=4
330  IF MOVE(D)=-1 GOTO 330
340  IF C=D THEN 390
350  ERA D
360  POSITION C,XPOS(D),YPOS(D)
370  D=C
380  MOVE C:GOTO 130
390  MOVE D:GOTO 130
400  IF S<>0 THEN 140
410  DEF SPRITE 0,(0,1,0,1,0)=
    CHR$(121)+CHR$(120)+CHR$(123)+
    CHR$(122)
420  DEF SPRITE 1,(0,1,0,1,0)=
    CHR$(165)+CHR$(164)+CHR$(167)+
    CHR$(166)
430  FOR N=1 TO 240

```



490 GOTO 140

はいけい つく 背景を作るBGグラフィックのパターン図





パート



キャラクタ と 自由に遊ぼう





マリオが 行ったり来たり

♣ ^フFOR^オ~^アNEXT^ネ命令は^ク ^{スト}同じ動作をくり返させる^{めい} ^{れい}

ここでは、FOR~NEXTの勉強をするわけだけど、FOR~NEXTはFORとNEXTの間をTOのあとに指定した数値の回数だけくり返すんだ。

```
FOR N=1 TO 10
```

```
    )          Nが1から10になるまで10回くり返す
```

```
NEXT
```

● 1から10までの数をタテに表示させる

ちょっとために、下のプログラムをRUNして、変数Nの増え方を見てみよう。

```
10 FOR N=1 TO 10  
20 PRINT, "N=" ; N  
30 NEXT
```

画面には1~10までの数字がタテに表示されたね。このような

働きをするのが、FOR~NEXTなんだ。

このFOR~NEXTの使い方は、同じことをさせるときには、とても便利な命令なんだよ。

●画面全体を青い■で塗りつぶす

たとえば、何かキャラクタを画面いっぱいに表示したい場合には、どうしようか？

```
10 FOR N=1 TO 700
20 PRINT "■";
30 NEXT
```

このプログラムをRUNすると、画面全体が青色になるよ。その意味は"■"をコンピュータが、いっしょうけんめいに書いているからなんだ。

●青と白のシマ模様を表示させよう

```
10 FOR N=1 TO 350
20 PRINT "■□";
30 NEXT
```

こんどは画面に青と白が代わるがわる出力されたから、画面にはシマ模様ができただろう。TO 350の数字をいろいろ変化させるとよく理解できるかもしれないな。



■の出し方は
カナモードで
GRPHキー
と¥ルキー
を押すんだよ

□の出し方は
カナモードで
GRPHキー
とへリキー
を押すんだよ



● ＊マークを左から右に動かそう

こんどはもう少しおもしろくしてみよう。キャラクタを左から右に走らせるプログラムだ。少し複雑だから、注意してね。

```
5  CLS
10  FOR N=1  TO  27
20  LOCATE N, 10
30  PRINT  "*"
40  FOR T=1  TO 100
50  NEXT
60  LOCATE N, 10
70  PRINT "  "
80  NEXT
```



書いて消して動いたようにみせる

これを RUN すると、＊マークが右に移動したはずだけど、どうしてかわかるかな？ ＊マークが動いたようにみえるけど、実は、最初に書いた＊マークを消すと同時に、その右に＊マークを書き、書いて消して、書いて消して……と同じ動作を何回もくり返して、動いたようにみせているんだ。



＊カウンターループ

40行と50行のFOR～NEXTは、カウンターループといって、少し時間をかけているところだよ。40行の100の数字を少なくすれば早く動き、多くすればゆっくりすすむよ。

＊書いて

＊消して書いて

＊消して書いて

＊消して書いて



＊動くようにみえる

●マリオをFOR～NEXTで右に動かす

ここでいよいよマリオの登場といこう。マリオを左から右へ移動させるには、MOVE命令のほかに、FOR～NEXTを使ってもいい。

```

10  CLS
20  SPRITE  ON
30  CGSET  1, 0
40  DEF  SPRITE  0, (0, 1, 0, 1, 0) =
    CHR$(1) + CHR$(0) + CHR$(3) + C
    HR$(2)
50  FOR  X=0  TO  244  STEP  4
60  SPRITE  0, X, 100
70  FOR  M=0  TO  100: NEXT
80  NEXT
90  END
    
```




左右逆のMarioが走るぞ！



このプログラムをRUNすると、右を向いたMarioが左から右方向へ走って行く。40行を見てみよう。ところがキャラクタテーブルにはどこをさがしても、右向きのMarioはないじゃないか。そのキャラクタを右に向かせて登録しているのがこの40行だよ。

SPRITE 0のあと、カッコ内4番目の数字が0なら左向きのそのままのキャラクタ表示、1なら右向き、つまり逆向きのキャラクタが表示できるんだ。でもこれだけではだめ。その後ろにあるCHR\$ の数字も逆に設定しなければならんだい。



*アニメキャラクタの表示

40行のDEF SPRITE 0は、表示するキャラクタ名を“0”とすることだが、これにつづく()の中の意味を教えよう。

(0, 1, 0, 1, 0)

アニメキャラクタを上下逆に表示したかったら
1

1は左右を逆にして表示。そのままの表示なら0

スプライト面のキャラクタがバックグラウンド面の文字の上に表示。1なら文字が上

4文字分の大きさで表示。0なら1文字分の大きさ

キャラクタテーブルと同じ配色


■キャラクタの逆立ちや裏返しの表示

キャラクタテーブルAにはいろいろなキャラクタがあって、どれでも自由に^{じゆう ひよう}に表示させることができるけど、このキャラクタを^{さか}逆立ち^だ（上下逆の表示）させたり、^{うらがえ}裏返し^き（左右逆の表示）させたりすることもできる。


前のプログラムの40行をいろいろ^か変えてためてみよう。

スロケラムの40行
DEF SPRITE 0,
(0,1,0,1,0)
=のあとを
いろいろ
変化させるのだ


(^{せいじよう ひよう}正常な表示) = CHR\$(0)+CHR\$(1)+
CHR\$(2)+CHR\$(3)




(^{さき ゆうぎやく ひよう}左右逆の表示) = CHR\$(1)+CHR\$(
(0)+CHR\$(3)+CHR\$(2)



(^{じゆう げぎやく ひよう}上下逆の表示)
= CHR\$(2)
+ CHR\$(3)
+ CHR\$(0)
+ CHR\$(1)



● 2つのキャラクタをドッキング




キャラクタテーブルAのマリオもレディもペンペンも、そのなかのキャラクタも、^{さか}逆立ちさせたり、^{ひだり}左右逆に表示させたり、^{じゆう}自由に表示させられるようになったはずだけど、この方法^{ほうほう}を応用^{おうよう}すれば、もっとおもしろいこともできそうだ。

たとえば、^{じょうはんしん}上半身がマリオで^{かはんしん}下半身がペンペンのヘンな生きもの^{ひょうじ}の^{ひょうじ}だ^{いじ}って表示させられるのだ。2つ^{いじ}以上のキャラクタをドッキングさせて、^{あら}新しいキャラクタを創造^{そうぞう}してみよう。

```
10 CLS
20 SPRITE ON
30 CGSET 1, 0
40 DEF SPRITE 0, (0, 1, 0, 1, 0) =
    CHR$(1)+CHR$(0)+CHR$(99)+CHR$(98)
50 FOR X=0 TO 244 STEP 4
60 SPRITE 0, X, 100
70 FOR M=0 TO 100: NEXT
80 NEXT
90 END
```

さあどうだ。^{じょうはんしん}上半身^{みぎ}は^{みぎ}右向きマリオ、^{かはん}下半身^{しん}は^{みぎ}右向きペンペンのヘンなキャラクタができたぞ。





数当てゲームで ファミコンに挑戦

♣ 文字変数も数値変数も 変数はいろいろ変わる数だ

ファミコンとチエくらべしてみようか。チエくらべとはちょっと大げさだけど、ファミコンが選んだ1～99までの数字を当てようっていう単純なゲームだ。ファミコンが選んだ数字を、何度目で当てられるかというわけ。

●ファミコンの考えている数字はナニ？

RUNしたあと「スウジハ」と表示されたら、コレダと思う数字をキー入力する。ただそれだけでいい。

きみがキー入力した数字が、ファミコンの選んだ数字より少なかったら、「モットオオイヨ」というヒントが表示される。ファミコンの数字より多かったときには「モットスクナイヨ」というヒントが出る。




```

10  CLS
20  A=RND (99)
30  KI=1
40  PRINT KI; "カイメ "
50  INPUT "スウジ ハ="; B
60  IF A=B THEN PRINT "ア タ リ"
    :GOTO 100
70  KI=KI+1
80  IF A>B THEN PRINT "モット オオ
    イヨ":GOTO 40
90  IF A<B THEN PRINT "モット スク
    ナイヨ":GOTO 40
100 PRINT "モウイチド シマスカ (Y/N)"
110 INPUT CY$
120 IF CY$="Y" THEN GOTO 10
130 END

```

このヒントをたよりに、^{すうじ}数字を入
^{なお}れ直して、さて何^{なん}度^ど目^めで^あ当^あた^ある^あか^あが
^{しょうぶ}勝負^{かい}だ。10回^{かい}も^{なお}やり^{なお}直^あして^あまだ^あ当^あ
 らない^あよう^あだと、き^あっ^あと^あファ^あミ^あコ^あン
 にバ^あカ^あに^あさ^あれ^あて^あし^あま^あう^あに^あち^あが^あい^あない^あ。

こ^あこ^あでは^あ遊^{あそ}び^{あそ}な^あが^あら^あ、^{へんすう}変^{へん}数^{すう}とい^あう
 む^あず^あか^あしい^あプ^あロ^あグ^あラ^あム^あィ^あン^あグ^あの^あテク^あニ
 ッ^あク^あを^あお^あぼ^あえ^ある^あの^あだ^あよ^あ。

クウゼンとカンがたよりだが
 けっこう頭^{あたま}を使^{つか}わなければ
 当^あたら^あない^あよ



変数はいろいろなものを入れる箱

プログラムの20行を見てみよう。

20 A=RND(99)

これは、Aという箱の中に、0～98までのどんな数でも1つ選んで入れなさいという文だ。いい換えれば、Aは、0～98までのどんな数にでも変わるよ、というわけで、このAが変数なのだ。

変数には2種類ある。数値変数と文字変数(ストリング変数)だ。上の例で、Aという変数は0～98までの数値に変わるのだから、これは数値変数だ。



* 数値変数の作り方

数値変数を作るときの約束ごとをおぼえよう。

1. 使えるのは、数値と英字だけ。
2. 1文字か2文字でつける。
3. 2文字のときの最初の文字と、1文字のときは必ず英字を使う。

(正しい例)

- A 1 英字と数値
- A 英字だけ
- A B 英字と英字

(まちがった例)

- 2 A 最初の文字が数値
- 1 数値だけ
- 2 3 数値だけ



数当てゲームのプログラムを読む

数値変数は、いろいろな数値を入れる箱だったね。それでは文字変数はいったいなんだろう？ いろいろな文字を入れる箱とでもいっておこうか。

文字変数については、あとでくわしく説明することにして、まず数当てゲームのプログラムを、最初からかんたんに説明しておこう。

10 CLS 表示させる画面をきれいにそうじだ。

20 A=RND (99) 数値変数Aに、0～98までの数値のうち、

なにかを代入する。ここでファミコンの選んだ数値が23だとすれば、変数Aには23が代入される。

30 KI=1 最初のゲームだから、変数KIには1を代入する。




40 PRINT KI;"カイメ";

50 INPUT "スウジ ハ=";B

いまのゲームが何回目を表示させたあと、きみたちがキー入力した数値を変数Bに入れる。

60 IF A=B THEN PRINT "ア タ リ":GOTO 100

ファミコンが考えた数値(A)と入力した数値(B)が同じだったら「アタリ」と表示してゲーム終了。100行に飛ぶ。



70 KI=KI+1

1回目^{かいめ}でまちがったひとは、回数^{かいすう}の変数^{へんすう} KI に+1する。

80 IF A>B THEN PRINT "モット オオイヨ":GOTO 40

入力^{にゅうりよく}した数値^{すうち} (B) が、ファミコンの選^{えら}んだ数値^{すうち} (A) より小^{ちい}さかったら、「モット オオイヨ」と表示^{ひょうじ}して40行^{ぎょう}へ。ゲームはやり直^{なお}した。

90 IF A<B THEN PRINT "モット スクナイヨ":GOTO 40



選^{えら}んだ数値^{すうち}より大^{おお}きかったら、「モット スクナイヨ」と表示^{ひょうじ}して40行^{ぎょう}へ。ゲームはやり直^{なお}し。

100 PRINT "モウイチド シマスカ(Y/N)"

画面^{がめん}に「モウイチド シマスカ(Y/N)」と表示^{ひょうじ}させる。

110 INPUT CY\$
120 IF CY\$="Y" THEN GOTO 10

もしYのキーが押^おされたら、10行^{ぎょう}にもどって、再^{また}びゲームができるようになる。



* IF~THEN

もし雨^{あめ}だったらファミコンしよう、というように、もし~^{じょうけん}だったら、と条件^{じょうけん}をつけるのがIF~THEN文^{ぶん}だ。


60行^{ぎょう}のIF A=B THEN...は、AとBが同^{おな}じだった

80行^{ぎょう}のIF A>Bは、AがBより大^{おお}きかったら

90行^{ぎょう}のIF A<Bは、AがBより小^{ちい}さかったらという意味^{いみ}だ。



出てくる数は まるででたらめだ

 ランダム RNDはサイコロの目だ
おも 思った数はなかなか出ない


ファミコンは、でたらめな数を、でたらめな順番で発生させたり表示させたりすることができるよ。

数当てゲームのプログラムにも、


20 A=RND(99)

というのがあったね。この意味は、0～98までの数のうち、どれでも勝手に変数Aに代入せよ、ということだった。

ここに使ったRNDが、でたらめな数をでたらめに発生させるものなのだ。ランダム関数っていうんだよ。

 でたらめな数を表示させる

RND (ランダム) 関数を使った短い短いプログラムで、でたらめな数の発生をたしかめてみよう。




```
10 A=RND(4)
20 PRINT A;
30 GOTO 10
```

このプログラムを RUN させてみよう。

0 から 3 までの数^{かず}が、でたらめ^{じゆんじよ}の順序^{が めん ひだり みぎ}で、画面^{びやうじ}の左から右にどんどん表示されていくね。

20 行^{ぎよう}の最後^{さいご}にある

; (セミコロン) は
次の文字^{つぎ もじ}の右どなりに
くっつけて表示^{ひやうじ}せよ
というシルシだ



1 3 0 2 3 3 | 0 0 2 3 | 2

2 0 3 0 0 | |



このプログラムは
お終^おわりがないから
止^とめたかったら
STOP キー
だよ



さっきは、でたらめ^{かす}な数^{かず}が、
画面^{が めん}の左^{ひだり}から右^{みぎ}へ、どん^{ひやう}どん表示^{ひやう}されていったけれど、こんど
は、画面^{が めん}の左^{ひだり}上^{うへ}から、どん^{ひやう}どん下^{した}に表示^{ひやうじ}されていくだけだね。

; (セミコロン) があるかな
いかで、こんなにちがうんだよ。

こんどは、20 行^{ぎよう}の最後^{さいご}にあ
る; (セミコロン) をつけなくて、

20 PRINT A

と、こうして RUN させてみよ
う。どうなったかな？

* RND 関数^{かんすう}

RND (n) というように
使^{つか}う。() 内^{ない}の n は 1 ~
32767 までの整数^{せいしゆう}だ。数式^{すうしき}も
使^{つか}えるよ。

RND (5) なら、0 ~ 4
までの数^{かず}、RND (10) な
ら、0 ~ 9 までの数^{かず}を発生^{はっせい}
させるんだ。



文字のたし算はできるのかな

✿ 文字変数は文字を入れる箱 入れたり出したりいそがしい

ステップ3では、数値変数^{すうちへんすう}っていうヘン^{へん}な数^{かず}について説明^{せつめい}したが、ここでは、もっとヘン^{へん}な数^{かず}を使^{つか}って遊^{あそ}んでみよう。このヘン^{へん}な数^{かず}は、文字変数^{もじへんすう}っていうんだ。

A\$ってなんだ？



これが文字変数^{もじへんすう}さ
\$ (ドルマーク)
をつけて
数値変数^{すうちへんすう}と区別^{くべつ}
してるんだ
それに も
だいじだぞ

数値変数^{すうちへんすう}は、とりあえず数^{かず}を入^いれたり出^だしたりする箱^{はこ}だとい^いって説明^{せつめい}したが、これ^こになら^なって、文字変数^{もじへんすう}は、文字^{もじ}や文字列^{もじれつ}を入^いれたり出^だしたりする箱^{はこ}だとい^いっておこう。

まずは、次^{つぎ}のプログラム^{じやうけい}を入^い力^{りき}して、RUN させてみよう。

文字変数を使って文字列を表示させよう

```
10 A$ = "ABCD"
20 B$ = "EFGH"
30 PRINT A$
40 PRINT B$
50 C$ = A$ + B$
60 PRINT C$
70 END
```

さて、画面にはどんな表示が

たかな？



・のなかが
表示されたぞ

```
ABCD
EFGH
ABCDEFGH
```



画面の上から「ABCD」、その下に「EFGH」、さらにその下に「ABCDEFGH」と表示されたはずだ。

プログラムをざっと見てほしい。10行にA\$、20行にB\$、50行にC\$という字が出てくるね。これを文字変数というのだ。

*新しいプログラムの入力

新しいプログラムを入力

するときは、ファミコンのメモリの中に古いプログラムが残っているとイケないから、まず念のため、NEW命令でメモリにあるプログラムを消去しよう。

古いプログラムが残っているかどうかたしかめたいなら、LIST命令で画面に表示させてみればいい。



10行の $A\$ = \text{"ABCD"}$ は、 $A\$$ という文字変数（文字を入れる箱に、“ ”（ダブルクォーテーション）で囲まれた ABCD という文字列を入れる（代入する）という意味だ。

30行の $\text{PRINT } A\$$ で、文字変数 $A\$$ に代入された ABCD を表示させている。

20行の $B\$ = \text{"EFGH"}$ も同じだ。40行の PRINT 文で表示される。



* 文字変数と変数値

文字変数は数値変数と区別するために、変数名のあとに \$（ドルマーク）をつける。

$A\$$ 、 $B\$$ 、 $C\$$ と、みんな \$ がついているね。

また、文字変数に入れるデータ（変数値）は、“ ”（ダブルクォーテーション）で囲む。
“ABCD” “EFGH” と、こういうぐあいだ。



50行の $C\$ = A\$ + B\$$ は文字列のたし算だ。 $C\$$ という文字変数に、 $A\$$ の ABCD と $B\$$ の EFGH を代入するわけだから、 $A\$$ と $B\$$ が並んだ ABCDEFGH が表示されている。

文字変数のたし算は
文字列と文字列を

くっつけて

並べるんだな



文字変数で

たし算はできるが

引き算、かけ算

割り算は

できないよ

自分の名まえを文字変数に入れる

こんどは、きみの名まえを入れてみようか。きみの名まえが、新
星太郎だとして、箭のプログラムをそっくりいただいてしまえ。

10 A\$ = "シンセイ"

20 B\$ = "タロウ"

30 PRINT A\$

40 PRINT B\$

50 C\$ = A\$ + B\$

60 PRINT C\$

70 END


シンセイ

タロウ

シンセイタロウ

いろいろなことばを入力して遊

んでみよう。






飛んで行ったり また戻ったり

♣ GOTOやGOSUBは 流れを変えるジャンプ命令

プログラムは、行番号の小さな順に実行していく、というタテマエはもう知っているね。だけど、例外はどこにでもあるものだ。GOTO 命令や GOSUB 命令もそのひとつだ。

行番号の小さい順から実行していくところか、そんなきまりは守ろうという気もさらさらなく、行きたい行番号のところに飛んで行ってしまふ。だから、GOTO や GOSUB 命令をジャンプ命令というんだよ。



＊が好きな数だけ画面をかざる

```

10  CLS
20  INPUT A
30  IF A>20 THEN 20
40  IF A=0 THEN 200
50  GOSUB 100
60  GOTO 20
100 FOR N=1 TO A
110 PRINT "＊";
120 NEXT
130 RETURN
200 END
    
```

20行に戻る

100行にジャンプ

サブルーチン

60行に戻る

このプログラムは、数字キーで
入力する20までの任意の数だけ、
画面に「＊」を表示させるんだ。

RUN させると、20行の INPUT
文でキー入力待ちになる。1 から
20までの数をキー入力しよう。

0 を入力するとプログラムは終
了だ。20以上の数を入力すると(30
行)、もう一度20行に戻って、ふた
たび入力待ちだ。

たとえば、5 を入力したとして、
プログラムを追ってみよう。

ここに出てくるジャンプ命令は
30行、40行の IF~THEN
50行の GOSUB
60行の GOTOだ



10 CLS ...画面^{がめん}をきれいにして、

20 INPUT A ...このAに5^{はい}が入る。

30 IF A>20 THEN 20 ...入力^{にゅうりょく}

た数が20以上だったら ~ という条件^{じょうけん}だから、この行^{ぎょう}は無視^{むし}。

40 IF A=0 THEN 200 ...入力^{にゅうりょく}

した数は0ではないから、この行^{ぎょう}も無視^{むし}。

50 GOSUB 100 ...100行^{ぎょう}へジャンプ

しろというから、100行^{ぎょう}に行^いってみよう。

ホイ、
きた

100 FOR N=1 TO A ...ここからサブルーチン^{しゅりょく}に入る。Aは5だから、次のこと^{つぎ}を5回^{かい}くり返す^{かえ}。

110 PRINT "＊"; ...「＊」を5回^{かいひょうじ}表示^{ひょうじ}するということになる。



120 NEXT ...100行^{ぎょう}のFOR文^{ぶん}を受けてFOR~NEXTの終了^{しゅうりょう}。

130 RETURN ...50行^{ぎょう}のGOSUB文^{ぶん}

を受けて、GOSUB~RETURNの終了^{しゅうりょう}。GOSUB文^{ぶん}の次の行^{ぎょう} (60行^{ぎょう})に戻る^{もど}る。

60 GOTO 20 ...また20行^{ぎょう}に戻^{もど}って、もう一度^{いっど}くり返^{かえ}し。



* サブルーチン

プログラムの主^{しゅ}となる流^{なが}れをメインルーチンというが、サブルーチン^{どくりつ}は独立^{どくりつ}した脇役^{わきやく}だ。

メインルーチン^{なか}の中に割^わ

り込んで、ひとつの役割^{やくわり}を果たしたあとで、またメインルーチン^{もど}に戻る。

このプログラムでは、100行^{ぎょう}からのFOR~NEXT文^{ぶん}がサブルーチンだよ。

■ ジャンプ命令はたくさんあるよ

※を表示させるプログラムには、ジャンプ命令がたくさん出てきたよ。どれがジャンプ命令だろう？

■ GOTO 命令

60 GOTO 20

20行に飛んで行け。


行番号の順序なんてほっといて、なにがなんでも20行にジャンプしてしまえ。



■ IF~THEN 命令

30 IF A > 20 THEN 20

もしAが20以上だったら、20行に飛んで行け。THENの後ろにGOTOを入れて、IF A > 20 THEN GOTO 20といっても同じことだ。




■ GOSUB~RETURN 命令

50 GOSUB 100

{

130 RETURN

まず100行にジャンプして、RETURN文に出会ったら前に戻って、GOSUB文の次の行に行けという命令だ。



● マリオが^{した}き^さどんどん下^さに下がっていく

^{まえ}前のプログラムで、GOTO や GOSUB~RETURN のジャンプ命令を^{めいれい}簡単に^{かんたん}説明^{せつめい}したけど、プログラムを^{じっこう}実行しても、ただ「*」が表示^{ひょうじ}されるだけで、ちっともおもしろくもなかったね。

こんどは、あのプログラムを^{おうよう}応用して、少しは^{へんか}変化のあるものを作^{つく}ってみた。キャラクターテーブルAにあるマリオのキャラクターを^{ひょうじ}表示させ、その^{ひょうじ}表示位置を、きみが操作^{そうさく}する数字^{すうじ}キーで^か変えてみようというわけだ。

```
10  CLS
20  SPRITE ON
30  INPUT A
40  IF A>255 THEN 30
50  IF A=0 THEN 130
60  GOSUB 100
70  GOTO 30
100 DEF MOVE (1)=SPRITE (0, 5, 1,
    A, 0, 0)
110 MOVE 1
120 RETURN
130 END
```



RUN するとすぐ^{じゅうりやく}入力^ま待ちの状態になる。そこで255^い以下の^{すうじ}数字^{ぶん}をキー入力^{にゅうりやく}してみる。押した^お数字^{すうじ}分^{ぶん}だけ^{した}下に、マリオが移動^{いどう}するよ。もし、0^{じゅうりやく}を入力すれば、プログラムは終了^{しゅうりょう}する。




左右に動く かわいいレディ

♣ キャラクタの表示や 移動も自由自在だよ

キャラクターテーブルA

キャラクターテーブルAのレディ

にあるかわいいレディの
キャラクタを左右に動か
すプログラムを作ってみ
た。次のプログラムを入
力してRUN させたら、
レディを動かすのは、コ
ントローラIの十字形ボ
タン ◀ と ▶ だよ。



● コントローラでレディを左右に動かす

```
10 CLS
20 SPRITE ON
30 CGSET 1, 0
```

```

40 DEF SPRITE 2, (2, 1, 0, 1, 0)=
   CHR$(33)+CHR$(32)+CHR$(35)+CHR
   $(34)
50 DEF SPRITE 3, (2, 1, 0, 0, 0)=
   CHR$(32)+CHR$(33)+CHR$(34)+CHR$(35)
60 SPRITE 2, 100, 140
70 X=100:Y=140
80 S=STICK(0)
90 IF S=0 THEN 80
100 IF S=1 THEN X=X+4:GOTO 200
110 IF S=2 THEN X=X-4:GOTO 210
120 GOTO 80
200 SPRITE 3:SPRITE 2,X,Y:G
   OTO 80
210 SPRITE 2:SPRITE 3,X,Y:G
   OTO 80

```



■ プログラムを簡単^{かんたん}に説明^{せつめい}しておこう

10~30行は初期^{しゅう}設定^{しやう}だ。キャラクタ^{うご}テーブル^{もんく}を動か^{うご}すときのきまり文句^{もんく}みたいなものだから、このまま暗記^{あんき}すればいい。

40行と50行は、表示^{ひょうじ}させるレディのキャラクタ^きを決^きめている。スプライト2のキャラクタ^{きゆう}は左右^{ざい}を逆^{ぎやく}にして右^{みぎ}を向^むいたレディ、スプライト3は、キャラクタ^きテーブルにあ

表示^{ひょうじ}させる
キャラクタ^きを
決めるのは
DEF SPRITE
だ






るままの左向きレディなのだ。

60行はSPRITE 2で定義した
右向きのレディを、X座標を100、
Y座標を140の位置に表示する
というわけだ。

* X座標とY座標

X座標は画面の横の位置、
Y座標は画面の縦の位置の
ことだ。

スプライト画面上の有効
表示範囲は、横(X座標)
0~240、縦(Y座標)5~
220だ。この範囲内だった
ら、レディは自由に動ける
が、この範囲外を指定する
とエラーになってしまうよ。



70行は最初に^{ざいしよ}表示させる座標^{ざひょう}位置。この位置を^い変化させてレディを^{さゆう}左右に動かすのだ。

80~120行が、コントローラの^{にゅうりよく}入力ルーチン。80行の S = STICK (0) で、コントローラ I の^{じゅうじけい}十字形ボタンのどの^{ほうこう}方向が押されたかを^{しら}調べる。

90行は、もし S に 0 が^{だいにゅう}代入されたらというわけで、ボタンが押されなかったら80行に戻る。

100行は、もし S に 1 (十字形ボタンの^{みぎほうこう}右方向) が^{だいにゅう}代入されたら^{みぎ}右に^{うご}動くといっている。

110行は、もし S に 2 (十字形のボタンの^{ひだりほうこう}左方向) が^{だいにゅう}代入されたら、^{ひだり}左に^{うご}動くといっている。

200行で^{みぎむき}右向きのレディ (SPRITE 2)、210行なら^{ひだりむき}左向きのレディ (SPRITE 3) を^{ひょうじ}表示させて、^{みぎ}右と^{ひだり}左に^{うご}動かす。


コントローラIIを
調べるときは
STICK(1)だ

* 十字形ボタン

キャラクタなどを^{うご}動かすとき、その^{ほうこう}方向を決めるのが十字形ボタンだ。

上方のボタン^{じゅうほう}↑を押せば^{うへ}上に、^{ひだり}左方のボタン^{うご}←を押せば^{なな}左に^{うご}動くが、^{なな}斜めに^{うご}動かすときは、たとえば、^{みぎ}右斜め下なら^{うご}→と^{うご}↓の^{ほうこう}方向を同時に押せばいい。

(十字形ボタン)



ボタンが押されたそれぞれ^{ほうこう}の方向には、ファミコンがおぼえている決まった^き数字^{すう}がついている。^{うへ}上なら8、^{した}下なら4だ。あとで^で出てくるから、おぼえておこう。






テレビの画面に ？が出たら

テレビの画面に？マークが表示されたら、ファミコンが「なにか入力してくれ」とお願いしているのだから、キーボードかあるいはコントローラを使って、なにか入力してあげなければいけない。

それでは、なにかを入力する命令には、どんなものがあるかな。少しくわしく説明しておこう。

ファミコンになにかを入力してやるときの命令には、INPUT、INKEY\$, STICK、STRIG などがある。これらの使い方をおぼえておこうか。少しおもしろ味がないかもしれないが、だいじなことだから、がまんするんだね。



ファミコンは声は出さないけど、かわりに？マークを出して、きみに話しかけるよ



イン ブ ッ ト にゅうりょく **INPUTで入力したら** プ リ ン ト ひょうじ **PRINTで表示させてみる**

つぎ
次のプログラムを入力して、RUNさせてみよう。


10 CLS

20 INPUT A RUN 

30 PRINT A 

40 END

?



なんとまあ短いプログラムだ。でも短くと
もりっぱなプログラムだ。どうなったかな？

画面の左上に？マークが表示されて、カーソルもあるね。これ
が「なにか入力してくれ」のサインなのだ。

ここでたとえば「8」を入力してみよう。画面にいま入力した
「8」が表示されるよ。

キーボードから
数値や数式でなく
BとかCとか
BCDとか
文字を入れたら
どうなる？

Oを入力したと
同じになって
Oが表示されるよ



だって、Aという
変数は、文字用じ
やなくて数値用の
入れものだからね

こんどはちょっと変えて、次のプログラムの実行だ。

```
10 CLS
20 INPUT A$
30 PRINT A$
40 END
```

Aに\$がついたな



こんどは、キーボードから文字を入れてみよう。EでもEFでもEFGでもいいよ。入力した文字がそのまま表示されたはずだ。

\$がつくと、
これは数値用ではなくて
文字用の入れものですよ、
という意味になるのさ



*数値変数と文字変数

```
20 INPUT A
30 PRINT A
```

というときのAは、数値変数といって、このAにはいろいろな数値や数式があてはめられる。

```
20 INPUT A$
30 PRINT A$
```

というときのA\$は、文字変数といって、いろいろな文字があてはめられる。

数値



文字


たとえば8と
数値を入力しても
いいよ
だけどこの8は
「8」という文字で
数値ではないよ

●メッセージをつけておくと便利だよ

ただ「？」が表示されるだけでは、なにを入力したらいいか、わからないかもしれない。それでは、もう少し親切なプログラムを実行してみよう。

```
10 CLS
20 INPUT "ナマエ ハ=" ; A$
30 PRINT A$
40 END
```

こんどは「ナマエ ハ=」と表示されて、キー入力の状態になるから、なにを入力すればいいか、すぐわかるね。新星太郎なら、「シンセイタロウ」と入力するわけだ。



* INPUT 命令の注意


1. メッセージをつけるときは、
" " で囲うこと、そして変数の
前に ; を忘れないこと。
2. 文字変数には , (コンマ)
は代入できないよ。 , は、「こ
こでキー入力終わり」という合図に
なってしまう。
3. キーボードからデータを入
力したら、かならず RETURN キ
ーを押す。

♣️ ^おどんなキーが押されたのか ^{ぜんぶ}全部わかってしまう^{インキーダラー}INKEY\$

キーボードから1文字^{もじ}だけ^{にやうりよく}を入力
 するのがINKEY\$^{めいれい}命令^{じぶん}だ。自分で
 ゲーム^{つく}を作ったりするときに、とて
 もたくさん^{つか}使うのが、このINKEY
 \$だから、よくおぼえておこう。

^{もじ}^{へんすう}文字変数 = INKEY\$

として、この文字変数^{もじ}は押されたキ
 ーの文字^{もじ}。そしてINKEY\$は、その
 文字^{もじ}がなにかを^{しら}調べるというわけだ。



● ^お押されたキーはなにかを^{しら}調べる

```

10  CLS
20  PRINT  "ナニカ キーヲ オシテクダサイ"
30  A$ = INKEY$
40  IF  A$ = "" THEN 30
50  IF  A$ = "E" THEN 80
60  PRINT  "イマ " ; A$ ; " ヲ オシマシタネ"
70  GOTO 20
80  PRINT  A$ ;
90  PRINT  " ハ ENDデス"
100 END
    
```



ナニカ キーヲ オシテクダサイ



イマ M ヲ オシマシタネ

ナニカ キーヲ オシテクダサイ



なにも^お押さないとい
つまでも
「オシテクダサイ」
と^で出ているよ

プログラムを RUN しよう。

「ナニカ キーヲ オシテクダサイ」

と^{ひょうじ}表示されただろう。キーを^お押してやろうじゃないか。

たとえば「M」を^お押すと——

「イマ M ヲ オシマシタネ」

と^{ひょうじ}表示されるだろう。どんなキーが^お押されたか、ファミコンはち
ゃんと知っているのだ。

こんどは「E」のキーを^お押してみよう。

「E ハ END デス」

と^{ひょうじ}表示されて、プログラムは^お終わりだ。

プログラムの読み方

書くほうも読むきみも、少しめんどくさいが、たまにはプログラムを少しくわしく説明してみよう。

10 C L S

20 PRINT “ナニカ キーヲ オシテクダサイ”

わかるね。画面をきれいして、まず「ナニカ キーヲ オシテクダサイ」というメッセージの表示だ。

* INKEY\$ はゲームでどんどん利用されている

ファミコンゲームにもいろいろあるね。ロールプレイングゲームからアドベンチャーゲーム、シミュレーションゲームやアクションゲームなど、たくさんの種類が出まわっているけれど、このプログラムの中でも、INKEY\$ はたくさん使われているんだ。

あるキーを押してビーム砲を射ったり、怪物に会ったキャラクタが逃げたり、地下密室のドアを開けたりなど、みんな INKEY\$ を利用しているんだと思うと、この命令もグッと親しみを感じさせるね。



30 A\$=INKEY\$


A\$という文字変数、この場合は、押されたキーの文字をA\$に代入している。

40 IF A\$ = "" THEN 30

なんのキーも押されていなかったら (A\$ = ""), 30行^{ぎょう}に行け。だから、キーが押されないかぎり、30行^{ぎょう}へ行って40行^{ぎょう}に行って、また30行^{ぎょう}に行つてと、グルグルまわりでキリがないよ。

50 IF A\$ = "E" THEN 80

押されたキーが「E」なら、80行^{ぎょう}へ行け。(そして80行^{ぎょう}に行けば、ENDということがわかるはずだ。)





```
60 PRINT "イマ " ; A$; " ヲ オシマシタネ"
70 GOTO 20
```

Eのキー^い以外のキー^がを押した人^{ひと}、たとえば「M」を押したなら「イマ M ヲ オシマシタネ」というメッセージ^うを受け取るはずだ。それでまた20行^{きう}に戻って^{もど}最初^{しよ}からやり直し^{なおし}。

```
80 PRINT A$;
90 PRINT " ハ ENDデス"
100 END
```

Eキーを押した人^{ひと}には「E ハ ENDデス」と表示^{ひょうじ}されて、プログラム終了^{しやうりやう}。



げきつい さくせん
●スターキラ一撃墜作戦

```
10 VIEW
20 SPRITE ON
30 DEF SPRITE 0, (2, 0, 0, 1, 0) = C
    HR$(213)
40 DEF SPRITE 1, (2, 1, 0, 0, 0) = C
    HR$(180) + CHR$(181) + CHR$(182) + C
    HR$(183)
50 DEF MOVE (0) = SPRITE (8, 7, 1,
    25, 0, 0)
60 POSITION 0, 250, 40
70 MOVE 0
80 A$ = INKEY$
90 IF A$ = "Z" THEN GOTO 200
100 IF MOVE (0) = 0 THEN MOVE 0
110 R = RND (20) + 1
120 GOTO 80
200 FOR N = 150 TO 0 STEP -1
210 FOR M = 1 TO R : NEXT
220 SPRITE 0, 120, N
230 IF N > 40 AND N < 50 THEN IF XP
    OS (0) < 129 XPOS (0) > 111 THEN 500
240 NEXT : GOTO 80
500 ERA 0
510 SPRITE 0
520 SPRITE 1, XPOS (0), YPOS (0)
530 END
```

プログラムを追加してみよう

RUN させて遊んでみよう。「Z」
のキーを押せば、スターキラーを撃
墜する弾丸が飛び出すんだ。けど、
少しヘンだと思わない？

スターシップを1回撃墜すると終
わってしまうね。このプログラムは
まだ、完成していないためなのだ。

1回だけの撃墜で終わらせないで、
つづけてしたい人は、530行からプロ
グラムをつづければいいということ
だ。

さあ、自分の力でプログラムを作
ってみようよ。



* 530行以下にプロ
ラムを追加するなら

530行以下、どんな内
要を加えていけばいい
のだろう？

たとえば、命中した
ときの得点やハイスコ
アを判断させたり、
10回打ったら終了にさ
せるとか……。

音を出させたり、あ
るいは BG グラフィ
ックで背景を描いたり
というのも、ゲームを
楽しくする要素だ。

●ゲームがもっと楽しくなるBGグラフィック●

背景にBGグラフィックで絵を描いておけば、画面がなんとなく楽しくなる。BGグラフィックは、このあとの章でくわしく説明するから、ここでは、ほんの参考程度に、BGグラフィックの背景パターン図を示しておこう。あいているところに、いろいろ描き加えてもいいよ。

■背景面を描くBGグラフィックのあるパターン図(スターキラー撃墜作戦)


	0	1	2	3	4	5	6	7	8	9	10	11	12	13	14	15	16	17	18	19	20	21	22	23	24	25	26	27
0																												
1																												
2																												
3																												
4																												
5																												
6																												
7																												
8																												
9																												
10																												
11																												
12																												
13																												
14																												
15																												
16																												
17																												
18																												
19																												
20																												
21																												
22																												
23																												
24																												
25																												
26																												
27																												

BGグラフィックを
知らない人は
使い方をおぼえてから
このパターン図を
応用してネ



126

BGグラフィックを
少し知っている人は
このパターン図で
背景を描くといひ



バトルアタック

ジャーン!

次なるプログラムは、ウォーゲームといこう。BGグラフィックの背景付きだ。

きみは戦車隊の隊長。@マークの味方歩兵が、強力な敵に囲まれて苦戦中。きみは、味方の歩兵を助け出さなければならない。

戦車を動かすのは
コントローラIの
十字形キーだよ



ゲームのルールと成功のヒント

- 敵の地雷は*マークだ。これに戦車が触れれば爆発して、ゲームは終了だ。
- 出発点は燃料補給地だ。ここに来ると燃料が補給できる。
- @マークを取るたびに、スコアは10点ずつ加算される。1ラウンドのゲームで、歩兵を10人助け出して燃料基地に戻れば成功で、1ラウンドの終了。再スタートだ。基地に戻ったとき燃料0でもゲームオーバーだ。
- 3ラウンド目からは、*マークの地雷が少し多くなり、5ラウンド目からは、さらに増加するから、細心の注意を払わなければ、自滅だ。
- 歩兵をたくさん助けるコツは、効率よく燃料を使いながら進むことだ。よく考えてコースを進まないと、燃料がなくなってしまうぞ。



■ BGグラフィックのパターン図

	0	1	2	3	4	5	6	7	8	9	10	11	12	13	14	15	16	17	18	19	20	21	22	23	24	25	26
0	A 10	A 10	A 10	A 10	A 10	A 10	A 10	A 10	A 10	A 10	A 10	A 10	A 10	A 10	A 10	A 10	A 10	A 10	A 10	A 10	A 10	A 10	A 10	A 10	A 10	A 10	A 10
	A 10				A 10												A 10							A 10			
	A 10				A 10												A 10							A 10			
	A 10				A 10				A 10								A 10							A 10			
	A 10				A 10				A 10							A 10	A 10	A 10	A 10				A 10				
5	A 10				A 10				A 10															A 10			
	A 10			A 10	A 10	A 10	A 10	A 10		A 10														A 10			
	A 10								A 10						A 10	A 10	A 10							A 10			
	A 10								A 10						A 10									A 10			
	A 10														A 10					A 10				A 10	A 10	A 10	A 10
10	A 10			A 10	A 10	A 10	A 10	A 10						A 10			A 10			A 10						A 11	A 10
	A 10								A 10					A 10			A 10			A 10						A 11	A 10
	A 10								A 10					A 10	A 10		A 10								A 10	A 10	A 10
	A 10								A 10					A 10											A 10		
	A 10	A 10	A 10	A 10		A 10			A 10																A 10		
15	A 10					A 10											A 10								A 10		
	A 10					A 10											A 10	A 10	A 10	A 10					A 10		
	A 10		A 10						A 10	A 10	A 10	A 10	A 10				A 10								A 10		
	A 10		A 10									A 10													A 10		
	A 10		A 10									A 10													A 10		
20	A 10	A 10	A 10	A 10	A 10	A 10	A 10	A 10	A 10	A 10	A 10	A 10	A 10	A 10	A 10	A 10	A 10	A 10	A 10	A 10	A 10	A 10	A 10	A 10	A 10	A 10	A 10

```


10  CLS
20  CGSET 1, 0
30  SPRITE ON
40  VIEW
50  DEF SPRITE 0, (0, 1, 0, 0, 0) = C
    HR $(120) + CHR $(121) + CHR $(122) + CH
    R $(123)
55  DEF SPRITE 1, (0, 1, 0, 0, 0) = C
    HR $(136) + CHR $(137) + CHR $(138) + CH
    R $(139)
60  DEF SPRITE 2, (0, 1, 0, 1, 0) = C
    HR $(121) + CHR $(120) + CHR $(123) + CH
    R $(122)
70  DEF SPRITE 3, (0, 1, 0, 0, 1) = C

```



```

HR$(138) + CHR$(139) + CHR$(136) + CH
R$(137)
72 T=0:PP=1
75 X=200:Y=100:SS=100:GG=20
76 SPRITE 0,X,Y
77 IF PP>=3 THEN GG=25
80 IF PP>=5 THEN GG=30
81 FOR N=1 TO GG
82 XX=RND(22)+1
83 YY=RND(19)+1
84 Z$=SCR$(XX,YY)
85 IF ASC(Z$)=1 THEN 82
86 IF N<=14 THEN LOCATE XX,Y
Y:PRINT "@"
87 IF N>=15 THEN LOCATE XX,Y
Y:PRINT "*"
90 NEXT
100 S=STICK(0)
110 IF S=0 THEN 100
120 IF S=1 THEN A=2:B=2
130 IF S=2 THEN A=0:B=0
140 IF S=4 THEN A=3:B=3
150 IF S=8 THEN A=1:B=1
160 IF S=1 THEN A=2
170 X1=X/8-1:Y1=Y/8-2
180 IF S=1 THEN A$=SCR$(X1+1,Y1)
190 IF S=1 THEN IF ASC(A$)=1
THEN 100
200 IF S=2 THEN A$=SCR$(X1-2,
Y1)
210 IF S=2 THEN IF ASC(A$)=1
THEN 100
220 IF S=4 THEN A$=SCR$(X1,Y1
+2)
230 IF S=4 THEN IF ASC(A$)=1
THEN 100
240 IF S=8 THEN A$=SCR$(X1,Y1
-1)
250 IF S=8 THEN IF ASC(A$)=1
THEN 100
    
```



```

260 IF S=1 THEN X=X+8
270 IF S=2 THEN X=X-8
280 IF S=4 THEN Y=Y+8
290 IF S=8 THEN Y=Y-8
293 SPRITE 0
294 SPRITE 1
295 SPRITE 2
296 SPRITE 3
300 SPRITE A, X, Y
310 SS=SS-1
315 LOCATE 5, 22:PRINT " "
320 LOCATE 5, 22:PRINT "ネンリョウ";SS
330 IF SS<=0 THEN 1000
340 IF (X=200 OR X=208) AND (Y=100
OR Y=108) THEN SS=100
350 V$=SCR$(X/8-1, Y/8-2)
360 IF V$="@" THEN LOCATE X/
8-1, Y/8-2:PRINT":T=T+10:GOTO 380
363 IF V$="*" THEN 1000
365 V$=SCR$(X/8-2, Y/8-2)
370 IF V$="@" THEN LOCATE X/
8-2, Y/8-2:PRINT":T=T+10
375 IF V$="*" THEN 1000
380 LOCATE 15, 22
390 PRINT " "
400 LOCATE 15, 22
410 PRINT "スコア=";T
420 IF (X=200 OR X=208) AND (Y=100
OR Y=108) THEN IF T>=100*PP TH
EN CLS:PP=PP+1:VIEW:GOTO 75
430 GOTO 100
1000 LOCATE 10, 22
1010 PRINT "GAME OVER"


```



パート

4

BG
グラフィック
で
絵を描いて





ステップ 1


絵のない ゲームなんて

♥ ^か描いて楽しい、^{たの}みて楽しい
^{ほい}きみだけの^{けい}背景画を^がつくる^つろう

ファミコンの画面には、バックグラウンド面とスプライト面の2つがあることは、もう、わかっているよね。

バックグラウンド (Back Ground) とは、背景^{はいけい}ということ。そして、BGグラフィックとは、背景画^{はいけいが}ということだ。

ゲームをするときだって、絵があれば、おもしろさやスリルだって、グーンとアップするし、「迷路^{めいろ}ゲーム」のように、BGグラフィックで迷路^{めいろ}を作^{つく}って、その中^{なか}を通^{とお}りぬけていくゲームもあるしね。BGグラフィックをおおいに活用^{かつよう}して、ファミコンをワイド^{つか}に使いこなそうよ。



ページ

BGグラフィックを起動させよう

BGグラフィックの起動方法には2通りあるから、きみのつごうのいほうを選んで始めよう。

ひとつは、電源を入れて、GAME BASICモード画面から起動させる方法。これは、BASICを選ぶときに、かならず通るところだから、もう、よく知っているね。

もうひとつは、プログラムを作ったりするBASICの画面から、GAME

BASICを呼び出して、BGグラフィックに移る方法だ。この呼び出し命令は「SYSTEM」。覚えておいてね。



* BGグラフィック

動かない背景を別のところで作って、ゲームそのものとドッキングさせて、より迫力あるゲームや画面にしようというもの。

このドッキング命令がVIEW命令だ。

* VIEW 命令

BGグラフィックは、画面上のバックグラウンド面に出力される。ベーシックのVIEW命令は、これをバックグラウンド面にコピーする命令だよ。

* SYSTEM 命令

プログラムを作成したりするBASICの画面から、GAME BASICを呼び出すための命令。

■おなじみ、この画面から



ベーシック
BASICでプログラム
さくせい
作成中なら

SYSTEM
でこの画面が出る

BASICにもどりたいなら、

ESC キーを押して、
次に **STOP** キー。

そうすると、やっぱりこの画面に戻るから、ここで**1**のキーを押せばBASICだ

GAME BASIC

- 1 BASIC
- 2 BG GRAPHIC
- 3 END

1, 2, 3 KEY IN !!


* RESET スイッチに注意

リセットスイッチを押して、はじめからやり直す方法もあるけど、一度リセットスイッチを押してしまったら、それまでBASICモードやBG GRAPHICモードで作っていたプログラムや背景画はみんなきれいになってしまうよ。リセットスイッチを押す前に、必要なものはSAVEしておこう。

GAME BASICモードで**3**のキーを押すと、これもリセットされてしまうから注意しよう。



♥ BGグラフィックの画面で どうなっているの？



このBGグラフィックの画面の使いかたについて知らない、
実際に絵を描くことができないから、くわしく説明しておこう。
画面にふった番号順に説明していくから、ひとつひとつしっかり
覚えていけば、あとはもう、自由自在に描けるようになるよ。

①カーソル

カーソルがあるところに、
グラフィックの図形1個を
表示するんだ。このカーソル
を移動させるのは、カー
ソル移動キー。上下左右、
どの方向もオーケー。好き
な位置に移動させて表示を
スタートさせよう。




②座標位置(X, Y)

座標なんていうとむずかしそうだけど、表示する場所はどこかということだ。




表示画面
の座標

ために、カーソル移動キーを押して、カーソルをどこかちがう場所に移してごらん。XとYの右よこに出る数字がかわっただろう？ この数値が、図形1個が表示される画面の位置なのだ。



Xがヨコ座標で、
Yがタテ座標。BG
グラフィックで描け
る画面の範囲は、X
が0～27で合計28マ
ス、Yが0～20で合
計21マス、28×21で、
合計588のマス目の
中だ。





⑥描くテクニックを示すモード

MODE (モード) というのは、
図形を画面に表示したり消したり
するテクニックの種類のことと考
えればいい。

今、SELECT モードになって
いるね。これは、右に一列に出て
いる図形を SELECT する(選ぶ)
モードということなんだ。

その図形の列の下で点滅してい
る、三角形の山のようなものがあ
るね。これがキャラクタ・マーカ
ーだ。



キャラクタ・マーカは、**INS** キーを
押すと左に移動し、**DEL** キーで右に移動
するから、好きな図形を選んでごらん。

選んだら、そこでスペースキーを押し
てみよう。ホラ、きみの選んだ図形が、
画面上のカーソルがあった位置に表示さ
れたね。


BGグラフィックのお絵描きは、キャ
ラクタ・マーカで図形を選んで、それ
を画面上の四角いカーソルの位置に移す
だけ。スペースキーを5~6回押してご
らん。ホラ、こうやって図形を組み立て
るのさ。

④ブロック遊びにも似た図形のもと

さっきから、図形と呼んでいるのがこのキャラクターたちだ。

このキャラクターを好きなように組み合わせ、画面というキャンバスに絵を描いていくのだが、ナ、ナント、キャラクターはぜんぶで104種類も用意されているのだ。

かなりのデザインのセンス、美的センスが要求されそうだが、これだけあれば、どんなおもしろい背景画だって描けるんじゃないかな？　どんどん、やってみよう。



A～Mまでの13グループにそれぞれ0～7までの8個のキャラクターだ



CLR/HOME

を押すとMグループのほうへ

SHIFT を押し

ながら

CLR/HOME

を押すとAグループのほうへ

変わるよ



⑤ マス目4個にひとつの色

MODE のとなりに表示されている「0」は、キャラクタにつける色の番号だ。色は0～3までの合計4色が用意されていて、色を指定するキーは **RETURN** キーだ。 **RETURN** キーを1回押すと、数字がかわるから、これもためしてみよう。

MODE 0
SELECT

ところで、この色つけで、とてもたいせつなことがあるんだ。

	0	1	2	3	4	5
0							
1							
2							
3							
4							
5							
...							

②で説明した座標位置は、表示される場所だけではなくて、表示される色にもおおいに関係があるんだよ。

ちょっと、左の座標図の一部を見て。色のついた4個のマス目、この4個はぜんぶ同じ色で表示される。この単位をカラーエリアというのだけど、この範囲の中で2色も3色もは使えないよ。



ためしに、この4つのマス目に、それぞれ色を変えてキャラクタを並べていってごらん。先に描いたキャラクタは、あとから描いたキャラクタの色に変わってしまったはずだよ。

ただし、いちばん上の段だけは、はんばだから、横2つのマス目がカラーエリアだよ。

* BGグラフィックとアニメキャラクタの合成

スプライト面^{めん}にはアニメキャラクタ^{ひょうじ}を表示し、バックグラウンド面^{めん}にはBGグラフィック^{ひょうじ}を表示するわけだけど、この2つの画面のX座標^{がめん}とY座標^{ざひょう}の関係は、少しちがう。

それぞれのX座標^{ざひょう}、Y座標^{ざひょう}にキャラクタ、または絵^えを表示して、そのままドッキングさせて表示^{ひょうじ}すると、ぶれてしまうことがあるから要注意だ。

背景^{はいけい}の絵^えにアニメキャラクタ^{ひょうじ}を合成^{ごうせい}させるときの座標^{ざひょう}は、次の式で割り出すんだよ。


ヨコ

スプライト面^{めん}(ヨコ) = (BG GRAPHIC(ヨコ) × 8) + 16

タテ

スプライト面^{めん}(タテ) = (BG GRAPHIC(タテ) × 8) + 16

こうして補正^{ほせい}して、プログラム上^{じょう}のXとYの値^{めい}を求めよう。





ステップ 2

テクニックを マスターしよう

♥ コマンドをフルに^{つか}使って きみもテクニシャンになろう

588個のマス^め目に、ひとつひとつキャラクタ^{ひょうじ}を表示させていくなんていうと、すごくたいへんでめんど^{きょう}うな作業^{さぎょう}みたいだけど、ところがどっこい、便利^{べんり}な描きかた^かのメニューが、いくつも用意^{ようい}されているから、安心^{あんしん}してね。

まず、**ESC** キーを押^おしてみよう。画面^{がめん}の左上^{ひょううえ}に、文字^{もじ}が出てきた。これが、BGグラフィックを描^かくためのテクニック・コマンド。ステップ1で説明^{せつめい}したMODE（モード）の種類^{しやうるい}だ。

絵^えを描^かいている途中^{ちゆうちう}でも、**ESC** キーを押^おと押^おせば、このモードリストは、いつでも出^でてきてくれるよ。

> SELECT
COPY
MOVE
CLEAR
FILE
CHAR

ESC キー、ポン、で
いつでも出^でてくるよ



●SELECT(セレクト)

キャラクタや絵を画面に表示する



B Gグラフィックを選んだときに、すぐに出てくる画面で使えるのがSELECTだったね。

SELECTには3つの機能があるよ。

1. はじめにいろいろためしたように、数多くのキャラクタの中から選んで画面に表示させる。

2. これから説明する、他のいろいろな機能を使って、描いたり消したり、くふうしたりして作った画面を、ひとつのまとまった画面として表示させる。

3. キャラクタを消す。アルファベットの[D]キーか、ファンクションの[F1]キーを押すと、カーソルの中のキャラクタが消える。



*消すときは色に注意

SELECTでキャラクタを消すとき、[F1]キーを使うと、同時に配色も変えてしまうから、注意してね。

COPY(コピー)

キャラクタをいくつもコピーする

お次は「COPY」を選んでみよう。画面下の MODE 表示の下は「COPY」となった。

COPY はコピーだよ。SELECT で同じキャラクタを1個1個表示させてもいいけど、もっと簡単な方法はないかな。というわけで、同じキャラクタをいくつもコピーしちゃうのだ。

いったんセレクトモードで描き込んだキャラクタの中で、いくつも使いたいキャラクタがあったら、その上にカーソルを重ねて **INS** キーを押す。

そのキャラクタを描き込みたい位置にカーソルを移動させて、**DEL** キーを押す。

また移動させて **DEL** キーを押す。またまた移動させて……いくつも同じキャラクタが描けたね。ね、便利だろう。これならラクチンチン、だ。

これはね、**INS** キーを押したときに、そのキャラクタがカーソルにコピーされて、カーソルがスタンプのような役わりにかわったからなんだ。スタンプなら、なん回押しても、同じものがでてくるわけさ。

▼▲でのマーカを動かしてメニューを選んだらスペースキーを押してセットだ




INSだ



DELだ



好きなだけ **DEL** を押そう



● MOVE(ムーブ)

キャラクタの引っ越しだ



「MOVE」は、「動かす」という意味。カーソルを重ねたところのキャラクタをそっくり別の位置に移動させるんだ。

COPY は、もとのキャラクタがそのままの位置に残るけど、MOVE は動かすちゃうわけだから、もとの場所にはキャラクタは残らない。キャラクタを別の位置に置いたら、もちろん、カーソルの中みもカラッポになる。



キーの役わりが変わるぞ

オイショと **INS** キー



DEL キーでお引っ越し

カーソルはスタンプから、引っ越し屋になるわけだ



* 頭は使いようだぞ

なにも表示されていないところにカーソルを移動して **INS** キーを押すと、空白のコピーだ。

このカーソルを消したいキャラタの上に重ねて **DEL** キーを押すと——そう、空白のスタンプを押しちゃったのだから、そのキャラクタは消えちゃう。

いちいち SELECT にもどって消さなくてもいい、COPY を使った修正テクニックだ。



●CLEAR(クリア)

消してもいいのかをよく考えて

マーカを「CLEAR」にあわせて、スペースキーを押したとたん、アッ——、絵がぜんぶ消えてしまった。もう一度、同じ絵を出すにはどうしたらいいのか？ かわいそうだけど、オシマイ。

このCLEARで画面を消してしまうと、再生することはできないのだ。うっかり使うと取り返しのつかないことになるよ。

一部分だけ消したいときは、SELECTに切りかえて、**[D]**キーを使って消したほうが安全だ。




●CHAR(キャラクタ)

グラフィック画面に文字を書く

SELECTやCOPY、MOVEで絵を描いているときには、キーボードにあるふつうの文字や数字、記号などは画面に描き込むことはできないのだけど、CHARならできるんだ。

直接、キーボードから打ち込めるから、いろいろためしてごらん。





描いた絵は とっておこうね

♥ テープに記録しておけば ゲームともすぐにドッキング

画面に描いた絵は、いったん電源を切るとあとかたもなく消えてなくなってしまうし、新しい絵を描きたいときは、どうしても画面をきれいにしなければならぬし、あとでゲームプログラムとドッキングさせたいから、今描いた絵は残しておきたいし。


まかせなさい、「FILE」です。

● FILE(ファイル)

カセットテープをデータレコーダにセットする。**[ESC]** キーでメニューを出したら、FILE を選んでスペースキーだ。

テープに記録するなら **[S]**
(SAVE)、テープから呼び出すなら **[L]** (LOAD) をキーボードから入力する。入力したら、**[RETURN]** キーを忘れずに。

SAVE (S), LOAD (L)? ■



次にファイルネームを聞いてくるから、16文字以内でつけて入力だ。

絵にあったタイトルをつけておくといいね。「メイロ」とか「ヤマ」とか……。

SAVE (S), LOAD (L) ? S

FILE NAME "

RETURN キーで
記録開始




♥ 座標もキャラクタ番号も色も まるごと紙に記録しちゃおう

こんどは、紙に記録しておく方法だ。こうしてとっておけば、いつでも、なん回でも利用できるよ。

なにを記録しておくかという、「SELECT」のところで説明した、「キャラクタテーブルB」と、画面の座標をまるごと記録しておくんだ。

この表のタテにふられているアルファベットがキャラクタグループ。ヨコにふられている数字が番号だ。



たとえば、「Fのグループの7番目」とい
 うように記録していくんだよ。そして、色
 もつけなければならぬから、0～3まで
 の配色番号を加える。

F 7 3

はいしよくばんごう
 配色番号

ばんごう
 キャラクタ番号
 キャラクタグループ

すると、こんなアルファベットと数字の
 組み合わせができるだろう。

これがひとつのキャラクタを表わすの
 だから、同じようにして、表示したキャラ
 クタと色を記録していけばいい。

記録するときには、まず紙にちゃんと画
 面の座標を作ること。入力するときに便利
 だし、ひと目でどんな絵かわかるだろう？



あまりうまくないけど
 サンプルを作ってみたから
 ためにしに画面に表示してみてね

	0	1	2	3	4	5	6	7	8	9	10	11	12	13	14	15	16	17	18	19	20	21	22	23	24	25	26	27
0																												
1																												
2																												
3																												
4																												
5																												
6																												
7																												
8																												
9																												
10																												
11																												
12																												
13																												
14																												
15																												
16																												
17																												
18																												
19																												
20																												
21																												
22																												
23																												
24																												
25																												
26																												
27																												



それ行け ゲームだ *Let's Key in!*

◆ 遊んだあとはオリジナル 負け^まないゲームを作ろうよ

ひととおりファミリーベーシックの勉強^{べんきょう}も終わ^おった。あとはきみの努力^{どりよく}しだいだ。わからなくなったら、へこたれないでなん回^{かい}でも読み返^よそうね。でも、本^{ほん}ばかりにらんでいたって、プログラミング^{しいうたつ}は上達^{じやうたつ}しないよ。どんどん key in することだ。

最後に、ゲームプログラムを2本^{ほん}載^のせておこう。

まず、正確^{せいかく}に key in すること。そして、遊^{あそ}んだあとで、プログラムをよ読んでみるといいよ。それほどむずかしいプログラムじゃないから、次^{つぎ}は、これらに負^まけないプログラム作^{つく}りに挑戦^{ちやうせん}だ。





スロットキャラクタ

◆ いっ ぱつ しょう ぶ 一発勝負のスロットマシン ファミコンでギャンブルだ

マリオやレディ、ペンペンやニタニタなど、ファミコンのスターたちがスロットマシンに^{かお}顔を出^だす。スリル^{まんてん}満点のファミコンギャンブルに^{ちようせん}挑戦^{とくてん}してもらおう。

● 得点キャラもあれば減点キャラもある

スロットマシンの3つの^{まど}窓に^{かお}顔を出^だすのは、マリオ、レディ、アキレス、ペンペン、ファイターフライ、ニタニタのおなじみのスターたちだ。

マリオ、レディ、アキレス、ペンペンは、きみたちももう知^しっているように、^{みぎ}右向^{ひだり}き、^む左向^{ある}き、歩^{ある}いていたり、6つのスタイルがあるけど、どのスタイルでもそろえばオーケーだ。

ファイターフライとニタニタは2つのスタイルで^で出てくるけど、この2つは^{げんてん}減点キャラクタだから、あまり^で出てほしくないね。

● あそ 遊びかた

ぐるぐるまわるスロットマシンをとめるのは、3つの^{すうじ}数字キーだ。向^{むか}かって左^{ひだり}の^{まど}窓が1、ま^{なか}ん中^{みぎ}が2、右^{みぎ}は^{まど}しの窓が3だ。

^{すうじ}数字キーを^①②③と^{じゅん}順^おに押^おしていくと、それぞれ^{じゅんぱん}順番^{かいてん}に回^{かいてん}転^{かいてん}がとまる。

②だけを^お押^おすと、左^{ひだり}の^{まど}窓とま^{なか}ん中^{まど}の窓がとまる。③を^お押^おすと、

3つの窓の回転がぜんぶ一度にとまるよ。

●ゲームのルール

1. 持ち点以上はかけられない。また、小数点の点数もかけられない。
2. 3つの窓に同じキャラクタがそろったらかけ点の10倍の得点になる。
3. 3つの窓がぜんぶそろわないときは、かけ点分だけ減点になる。
4. 左から2つそろえると、かけ点の5倍の得点になる。
5. ファイターフライかニタニタが、1と2の窓に2つそろってしまうと、かけ点の10倍の減点になる。
6. ファイターフライかニタニタが、1の窓に出たら、かけ点の5倍の減点になる。
7. ニタニタがまん中の窓に出ると、かけ点の5倍の減点になる。



けっこうシビアなスロットマシンだけど、それだけにおもしろいはずだよ。減^{げん}点キャラのニタニタなんか、ニタニタ～なんて顔を出してら、ホントに頭^{かま}にきてしまって、ますますアツくなっちゃうね。

●プログラムとBGグラフィックについて

ゲームを動か^{うご}かすためのプログラムは10行から410行。1000行から1120行、2000行から2030行は画面に文字を表示するためのプログラムだ。もちろん、続けて key in して、行番号^{ぎょうばんごう}の小さい順から実行^{こうぎん}されていくわけだけど、こういうプログラムの書きかたも覚えておくと便利だよ。

独立^{どくりつ}したブロックは、行番号^{ぎょうばんごう}をとばして、たとえば1000行からとか2000行とかきりのいい行番号^{ぎょうばんごう}にして組み込んでおくと、ひと目でそのブロックのプログラムがなんのためのプログラムかがわかるからね。もちろん、メインルーチンの飛び先^{と きさき}などには注意^{ちゅうい}してね。

BGグラフィックは、もっと凝^こって見たければ、どんどん描^かき加^くえたり、色^{いろ}を変^かえたりしていいよ。ただし、キャラクタが表^{ひょう}示^しされる窓^{まど}はプログラムで設定^{せってい}してあるから、窓の部分^{まど}を移動^{いどう}してもらおうと困^{こま}るけどね。



■ ゲームプログラム

スロットキャラクタ

```

10  CLS
20  CGSET 1, 0
30  SPRITE ON:VIEW
40  S=100
50  GOSUB 1000
60  A1=RND (26) * 4
70  DEF SPRITE 0,(0, 1, 1, 0, 0)=CHR$(A
    1)+CHR$(A1+1)+CHR$(A1+2)+CHR$(A
    1+3)
80  A2=RND (26) * 4
90  DEF SPRITE 1,(0, 1, 1, 0, 0)=CHR$(A
    2)+CHR$(A2+1)+CHR$(A2+2)+CHR$(A
    2+3)
100 A3=RND (26) * 4
110 DEF SPRITE 2,(0, 1, 1, 0, 0)=CHR$(A
    3)+CHR$(A3+1)+CHR$(A3+2)+CHR$(A
    3+3)
120 SPRITE 0, 40, 80
130 SPRITE 1, 64, 80
140 SPRITE 2, 88, 80
150 ST$=INKEY$
160 IF ST$="1" THEN T1=1
170 IF ST$="2" THEN T1=2
180 IF ST$="3" THEN T1=3
190 IF T1=1 THEN 80
200 IF T1=2 THEN 100
210 IF T1=3 THEN 300
220 GOTO 60
300 IF (A1>=0 AND A1<=24) AND (A2
    >=0 AND A2<=24) AND (A3>=0 AN
    D A3<=24) THEN S=S+C*10:GOTO 50
310 IF (A1>=0 AND A1<=24) AND (A2
    >=0 AND A2<=24) THEN S=S+C*5:

```

```

      GOTO 50
320  IF (A1 >=28 AND A1 <=52) AND (A2
      >=28 AND A2 <=52) AND (A3 >=28 AN
      D A3 <=52) THEN S=S+C*10:GOTO 50
330  IF (A1 >=28 AND A1 <=52) AND (A2
      >=28 AND A2 <=52) THEN S=S+C*5:
      GOTO 50
340  IF (A1 >=64 AND A1 <=84) AND (A2
      >=64 AND A2 <=84) AND (A3 >=64 AN
      D A3 <=84) THEN S=S+C*10:GOTO 50
350  IF (A1 >=64 AND A1 <=84) AND (A2
      >=64 AND A2 <=84) THEN S=S+C*5:
      GOTO 50
360  IF (A1 >=96 AND A1 <=108) AND (A2
      >=96 AND A2 <=108) AND (A3 >=96 AN
      D A3 <=108) THEN S=S+C*10:GOTO 50
370  IF (A1 >=96 AND A1 <=108) AND (A2
      >=96 AND A2 <=108) THEN S=S+C*5:
      GOTO 50
380  IF (A1 >=56 AND A1 <=60) AND (A2 >=56
      AND A2 <=60) THEN S=S-C*10:GOTO 50
385  IF (A1 >=88 AND A1 <=92) AND (A2 >
      =88 AND A2 <=92) THEN S=S-C*10:G
      OTO 50
390  IF (A1 >=56 AND A1 <=60) OR (A1 >
      =88 AND A1 <=92) THEN S=S-C*5:G
      OTO 50
395  IF (A2 >=88 AND A2 <=92) THEN S=
      S-C*5:GOTO 50
397  IF (A1 >=56 AND A1 <=60) AND (A2
      >=88 AND A2 <=92) THEN S=S-C*10:
      GOTO 50
400  S=S-C:T1=0
410  GOTO 50

```

```


1000 LOCATE 18, 3:PRINT "
1010 LOCATE 13, 3
1020 PRINT "モチ コイン ハ";S
1030 IF S<=0 THEN GOTO 2000
1040 LOCATE 19, 10
1050 PRINT "コイン ヲ"
1060 LOCATE 20, 11
1070 PRINT "カケテ クダサイ"
1080 LOCATE 22, 13
1090 INPUT C
1100 IF C>S THEN 1000
1110 T1=0
1120 RETURN
2000 CLS
2010 LOCATE 13, 3
2020 PRINT "GAME OVER"
2030 END
    
```



■スロットキャラクタのBGグラフィック

	0	1	2	3	4	5	6	7	8	9	10	11	12	13	14	15	16
0																	
1																	
2																	
3																	
4																	
5																	
6																	
7																	
8																	
9																	
10																	
11																	
12																	
13																	
14																	
15																	
16																	
17																	
18																	
19																	
20																	





マスターマインド

◆ 4つの数の組み合わせ きみのカンがためされるぞ

^{あそ}遊びかたはかんたんだ。ファミコンがかくし^も持っている4つの^{かず}数をあてるゲームだ。AからDまでの4つの^{はこ}箱に、コレカナ?と^{おも}思った^{かず}数を入れて^いくだけ。1から9までのうち4つの^{すうじ}数字を^く組み^あ合わせると……これはもう、^め目がまわるくらいあるぞ。

ゲームプログラム

マスターマインド

```
10  CLS
20  CGSET 1, 0
30  VIEW
40  DIM A (8)
50  E = 1
60  FOR I = 0 TO 8
70  K = RND (9)
80  A (I) = K
90  IF K = 0 THEN 70
100 IF I = 0 THEN 140
110 FOR J = 0 TO I - 1
120 IF A (J) = K THEN 70
130 NEXT
140 NEXT
200 IF A (0) = 0 OR A (1) = 0 THEN 50
210 IF A (2) = 0 OR A (3) = 0 THEN 50
220 A = A (0)
```

```

230 B=A (1)
240 C=A (2)
250 D=A (3)
260 LOCATE 9, 22
270 INPUT "A="; A1
280 LOCATE 4, 7
290 PRINT A1
300 LOCATE 9, 22
310 INPUT "B="; A2
330 LOCATE 9, 7
340 PRINT A2
350 LOCATE 9, 22
360 INPUT "C="; A3
370 LOCATE 14, 7
380 PRINT A3
390 LOCATE 9, 22
400 INPUT "D="; A4
410 LOCATE 19, 7
420 PRINT A4
430 LOCATE 9, 22
440 PRINT "      "
450 IF A=A1 AND B=A2 AND C=A3 AND
D D=A4 THEN 500
460 LOCATE 14, 16
470 PRINT E; " カイメ デス"
480 E=E+1
490 GOTO 260
500 LOCATE 4, 22
510 PRINT " オ オ ア タ リ"
520 LOCATE 5, 22
530 PRINT "モウイチドスルトキハ"
540 LOCATE 8, 23
550 PRINT " RUNヲシテクダサイ。"
560 END

```

■マスターマインドのBGグラフィック

0 1 2 3 4 5 6 7 8 9 10 11 12 13 14 15 16 17 18 19 20 21 22 23

0
1
2
3
4
5
6
7
8
9
10
11
12
13
14
15
16
17
18
19
20

[illegible]

★ファミリーコンピュータ・ファミコン、
ファミリーベーシックは任天堂の商標です。

この本の内容についての問い合わせは、
株式会社 新星出版社 ☎03-831-0743
ファミリーコンピュータ係までお願い
いたします。

ファミリーコンピュータ™

絵でわかるファミコンベーシック

1986年8月15日 初版発行 ©

著 者	山 下 利 秋
発 行 者	富 永 弘 一
印 刷 所	慶昌堂印刷株式会社

発行所	東京都台東区 株式会社 新星出版社 台東2丁目24
郵便番号	110 電話(831)0743 振替東京4-72233

ISBN4-405-06072-X

任天堂

TM

TM

ファミリーコンピュータ

£

ファミリーベーシック

▶ 思いっきり遊ぼうぜ！ ファミコン-WORLD ◀

THE HYRULE FANTASY

ゼルダの伝説™

必勝攻略ガイド

地上冒険マップつき 定価350円

☆レベル別クリアガイド、敵モン
スター攻略法、ひみつをあかす
ヒントがぎっしりの決定版！☆



ファミコンゲーム 必勝攻略法

定価650円

スーパーマリオブラザーズ
ドルアーガの塔
スターフォース/スパルタンX

逆転必勝！これが噂の☆Vテク☆でい！

ぼくらのファミコン入門

定価780円

作れるカナ？君だけのオリジナルゲーム！

絵でわかる

ファミコンベーシック

定価680円

ゲームじゃ物足りない！待望のベーシック第2弾！



面白すぎるゲーム /
ファミコン-WORLD

ファミコン

ゲームの世界

ファミコンゲーム

ファミコン入門

ファミコンゲーム



ファミコンベーシック

絵でわかるファミコンベーシック

★
新星出版社
385



絵でわかる ファミコンベーシック

★ 新星出版社

定価680円

●ファミリーコンピュータ・ファミコン、
ファミリーベーシックは任天堂の商標です

ISBN4-405-06072-X C2055 ¥680E

ファミコンベーシック

ぼくらのファミコン入門

ファミリーベーシックを覚えてゲーム作り。キーボードの操作法とベーシックによるプログラミングの基礎をわかりやすく説明した。

B6判
定価780円

ファミリー コンピュータ™
ファミリー ベーシック™

絵でわかる
ファミコンベーシック

山下利秋

新星出版社

ニュー・コーポレーション
ニュー・コーポレーション

絵でわかる

スモールビジネス



新星出版社

385

絵でわかる
ファミコンベーシック

★ 新星出版社

●ファミリーコンピュータ・ファミコン、
ファミリーベーシックは任天堂の商標です